
Abstract

Generating functions (i.e. power series) have applications throughout enumer-

ative and analytic combinatorics. In this thesis we present Genfunlib, a new

Mathematica package containing a selection of implementations of symbolic

methods related to generating functions. With Genfunlib one can find the gen-

erating functions for regular languages, compute the initial terms of a generating

function, convert between generating function equations and recurrences, and

find asymptotics. This thesis gives mathematical background, extensive docu-

mentation for Genfunlib, and tutorials.

1 Introduction

According to the NSF, a scientist’s worth is not dependent solely

on publications. . . . In the next five years, I believe that it will be-

come routine to track—and to value—citations to an online lab note-

book, contributions to a software library, bookmarks to data sets from

content–sharing sites such as Pinterest and Delicious.

– Heather Piwowar, 2013 [27]

This thesis is on Genfunlib, a package for the Mathematica computer algebra

system, which was written by the author of this document. The goal of Gen-

funlib is to provide commands for obtaining and using generating functions to

solve combinatorial problems. The full source code of Genfunlib can be found

in the Appendix; it is also available on the web at this address:

https://github.com/amacfie/Genfunlib

Genfunlib consists of implementations of a number of well–known proce-

dures not currently available in Mathematica or any third–party Mathematica

package. Genfunlib includes comprehensive finite automata functionality that

allows the user to convert between automata and regular expressions, and per-

form standard operations such as union, intersection, and reversal. Generating

functions that count the number of words in a given regular language can be

computed. These generating functions can be useful for counting problems that

involve regular languages such as pattern avoidance in words. Genfunlib uses

known methods to produce asymptotic information about a counting sequence

given its generating function. From a power series equation, Genfunlib can

compute the initial terms of the solution series. Genfunlib improves on Math-

ematica’s built–in capabilities for converting between recurrence relations and

power series equations. The symbolic method for obtaining generating functions

from constructible combinatorial classes is also implemented in Genfunlib.

The contents of this thesis are as follows. In Section 2, we give an overview

of enumerative combinatorics, generating functions, and their applications to

analysis of algorithms. Thanks to this section, the mathematical content of this

thesis is fairly self–contained. Section 3 is a brief overview of Mathematica pro-

gramming to allow readers to better understand the code in this thesis. Section

4 goes through each of Genfunlib’s commands and explains how to use them,

what they do and, where educative, how they do it. It is divided into subsections

that cover all the commands related to one mathematical topic. This section

can be considered to be both user and developer documentation. Boxes labeled

“Mathematica note” can be ignored by readers unfamiliar with Mathematica. In

Section 5 we present a number of relatively elementary combinatorial problems,

and solutions that make use of Genfunlib and Mathematica. In the solutions,

we attempt to give a sense of the range of Genfunlib’s functionality and try

to minimize the number of operations performed “by hand”. The problems in

Section 5.1 come from high school mathematics contests, or contest preparation

material. Finally, in Section 6 we conclude by offering possible paths for further

work.

Naturally, Genfunlib does not implement all known algorithms that one may

want to use with generating functions. Areas for expansion and improvement

are discussed as they are relevant at various points in the body of the thesis.

Further, there are tests and operations on power series for which no algorithm

has been found, e.g. whether a given univariate rational function has only pos-

itive coefficients [16].

2 Mathematics for the analysis of algorithms

In theoretical computer science, the greatest innovation is the realisa-

tion that algorithms are mathematical objects, and can be rigorously

analysed in terms of their consumption of scarce resources, including

space, time, and randomness.

– Jeffrey Shallit, 2006 [33]

Given a computer program, one may desire to know how long it will take to

run in absolute terms, that is, in physical units of time. Indeed ultimately,

such concrete predictions are always the end goal, and engineering methods

involving statistics, simulation, and detailed knowledge of particular computing

environments are certainly used (when rigorous this is known as the experimental

analysis of algorithms). However, given the complexity of actual computers, a

more abstract problem is taken to be the typical one in the field of computer

science called analysis of algorithms.

First, instead of executable computer programs, the objects of study are

algorithms [13], generally given at a level of abstraction where pseudocode is

sufficient. The physical running time of an algorithm is not defined, but the

number of steps it takes is. It is assumed that the number of steps taken

by an algorithm gives information related to actual running time of computer

implementations. From now on, the “running time” of an algorithm means the

number of steps it takes.

Second, because algorithms are abstract objects and do not run on comput-

ers, the goal is not to predict absolute timings, but to compare algorithms with

each other. Let T (A, i) be the running time of an algorithm A on an input i.

Let I(n) be the set of inputs with size n. We can represent the running time of

3



A with respect to a probability mass function fn : I(n)→ [0, 1] by the sequence

(TA(n)), where TA(n) =
∑
i∈I(n)

T (A, i)fn(i).

That is, TA(n) is the expected running time of A according to the probability

distribution defined by fn. The most common distribution is “worst–case”, which

assigns probability 1 to an input in I(n) on which the running time of A is

maximal. Another common distribution used is “average–case”, where

fn(i) = 1/|I(n)|,

for all i. More nuanced analysis has been done beyond these two distributions,

notably smoothed analysis and analysis using the Solomonoff–Levin measure

[19].

Once the distribution has been chosen, for two algorithms A and B com-

puting the same function, we have the two sequences (TA(n)) and (TB(n)).

Obviously, if TA(n) ≥ TB(n) for all n, it is easy to compare A and B, but this

ordering is not always enough. Instead, analysis of algorithms uses asymptotics:

relations are established between TA(n) and TB(n) as n approaches infinity.

Now, instead of directly comparing TA(n) to TX(n) for each X, we can, analo-

gously to the “interlingua” approach to machine translation [15], find a relation

between TA(n) and a simple standard sequence, after which, to compare algo-

rithms we need only compare simple standard sequences, which is easy.

Definition 1 (Asymptotic notation). As n→∞,

• f(n) = O(g(n)) means ∃c > 0, N > 0 : ∀n, n > N =⇒ f(n) < cg(n),

• f(n) = Ω(g(n)) means g(n) = O(f(n)),

• f(n) = Θ(g(n)) means f(n) = O(g(n)) and g(n) = O(f(n)),

4



• f(n) = o(g(n)) means limn→∞
g(n)
f(n) =∞,

• f(n) ∼ g(n) means limn→∞
g(n)
f(n) = 1, and

• f(n) = ω(g(n)) means limn→∞
g(n)
f(n) = 0.

Example 1. Suppose TA(n) =
∑n
k=1

n
k , TB(n) = 2n, and TC(n) = 1

n+1

(
2n
n

)
.

Then we have, as n→∞,

TA(n) = Θ(n log(n)),

TB(n) = Θ(2n),

TC(n) = Θ

(
4n

n3/2

)
.

By comparing n log(n), 2n, and 4n

n3/2 , we see TA(n) = o(TB(n)) and TB(n) =

o(TC(n)), as n→∞.

Asymptotics in this setting are perhaps surprisingly useful. The obvious

objection to their use is that, for example, even though (1 + exp(−1010))n is in

every asymptotic sense greater than n100, the former is in every practical sense

less than the latter. It is simply an observed fact that asymptotic statements

are usually practically relevant, even though so much information is lost.

Thus, we see that the prototypical methodology in analysis of algorithms

is the (incomputable in general) passage from algorithms to running time se-

quences to asymptotic expressions. (As always in mathematics, more abstract

problems may be solved in analysis of algorithms which support this methodol-

ogy indirectly.) This progression can be broken down into its two stages.

Obtaining exact running time sequences is closely related to the field of

mathematics called enumerative combinatorics, a.k.a. counting. For example,

let Sn be the highest–cardinality set of states that an algorithm A enters on

an input of size n. Then |Sn| is the worst–case running time of A on inputs

of size n. Counting sets such as Sn is precisely the field of enumerative combi-

natorics. According to Wilf [39], when an algorithm is found that takes n and

returns |Sn| and runs in at most exponential time, the combinatorial counting

problem associated with the sets Sn has been solved. To obtain these cardi-

nality algorithms, a set of counting strategies/techniques is used, including the

principle of inclusion–exclusion, recurrences, refinements, decompositions, and

Pólya theory.

Instead of passing directly from a set to its cardinality, in enumerative com-

binatorics, establishing a bijection between two sets is a very common goal. If

something has been proved about the cardinality of a set S1, which is shown to

be in bijection with S2, the same result applies to S2.

Problems in enumerative combinatorics may come from places other than

analysis of algorithms, such as combinatorial number theory, algebra, and graph

theory.

Given a sequence (fn), we define the ordinary generating function of (fn) to

be

F (z) =
∑
n≥0

fnz
n,

and we define the exponential generating function of (fn) to be

F̂ (z) =
∑
n≥0

fn
zn

n!
.

It turns out that these power series representations can be very useful for solv-

ing combinatorial problems, and especially useful for obtaining an asymptotic

estimate of a sequence.

We define two common classes of power series. A power series f(z) ∈ Q[[z]]

is algebraic iff there exist p0(z), . . . , pj(z) ∈ Q[z] such that

j∑
k=0

pk(z)f(z)k = 0.

A power series f(z) ∈ Q[[z]] is holonomic iff there exist p0(z), . . . , pj(z) ∈ Q[z]

such that
j∑

k=0

pk(z)f(z)(k) = 0,

where f(z)(k) is the kth formal derivative of f(z). Algebraic power series are

holonomic [36].

3 A few notes on Mathematica

In order to fully understand all of the Mathematica code used in this thesis,

the reader would have to be quite familiar with the Mathematica programming

language. Ruskeepää’s book [32] is ample preparation, for example. However,

readers who are not familiar with Mathematica should still be able to follow

the important ideas, and this section provides some essential Mathematica in-

formation to help in this regard.

The Mathematica code in this thesis is shown as it appears in an interactive

session: There are input lines, which are typed in by the user after the prompt

In[k]:= , and output lines, which Mathematica returns. Input lines that end

in a semicolon do not produce an output line. Here is a basic example:

In[1]:= 2 + 2

Out[1]= 4

In[2]:= 3 + 3

7



Out[2]= 6

Every input line is numbered and its output line has the same number.

The character % stands for the expression in the previous output line.

In[3]:= 4 - 2

Out[3]= 2

In[4]:= % - 2

Out[4]= 0

In Mathematica, functions take their arguments in square brackets.

In[5]:= Sin[Pi]

Out[5]= 0

To write an anonymous function, write the value in terms of the argument #,

and place an & at the end. In the next snippet, we define a function f that

squares its argument.

In[6]:= f = #^2 &;

In[7]:= f[4]

Out[7]= 16

Some of Mathematica’s built–in functions use assumptions to control their

behavior. They work like this:

In[8]:= FullSimplify[(-1)^(2a)]

Out[8]= (-1)^(2a)

In[9]:= FullSimplify[(-1)^(2a), Assumptions ->

Element[a, Integers]]

Out[9]= 1

In In[9] we pass an Assumptions option to the algebraic simplifier function

FullSimplify that says a can be assumed to be an integer. Alternatively,

assumptions can be specified by setting the global $Assumptions variable:

In[10]:= $Assumptions = Element[a, Integers];

In[11]:= FullSimplify[(-1)^(2a)]

Out[11]:= 1

Lists in Mathematica are made with braces and commas:

In[12]:= list = {5, 4, 3, 2, 1, 0};

In[13]:= Length[list]

Out[13]= 6

Mathematica lists can contain anything:

In[14]:= l = {5, "this is a string", {"string in a nested list!",

True, symbol}};

It is perfectly fine for a function to not evaluate:

In[15]:= Sin[2]

Out[15]= Sin[2]

If we want the numerical value of sin(2), for example, we pass it to the N function:

In[16]:= N[Sin[2]]

Out[16]= 0.909297

Mathematica has a huge number of built–in functions. To learn about a

built–in function, simply search the online documentation [29].

4 Guide to Genfunlib

4.1 Power series and their coefficients

Converting between power series and their sequences of coefficients is a basic

operation in computer algebra. In general, power series are represented by

equations that they satisfy, which may be differential equations. This includes

equations implied by “closed–form” expressions. Analogously, sequences may be

given by recurrence relations or closed–form expressions.

It is tempting to say that when performing the conversion between power

series and coefficient sequences, closed–form power series equations should be-

come closed–form expressions for coefficients, and nontrivial power series equa-

tions should become recurrence relations. However, while it generally can, why

should this aspect of the structure of an equation be preserved? Why not trans-

form

F (z) =
1

1− 2z

to

fn = 2fn−1, f0 = 1 instead of fn = 2n?

Indeed, if one wants a recurrence relation for computing initial terms, for com-

putational efficiency reasons it is preferrable to have a recurrence relation that

includes a finite set of lower-index terms, as opposed to one that uses the full-

history of the sequence. Thus, unless the form that the result should be in is

explicitly stated, there are many ways the translation can be performed [23].

Mathematica 9 includes some functionality for converting between power

series and their coefficient sequences; Genfunlib extends this functionality by

increasing the set of power series equations/recurrence relations on which the

transformation can be applied.

Mathematica note: In computer algebra, representing a power series

by an equation it satisfies is useful because of the level of generality it affords.

Mathematica has support for two kinds of power series equation objects that

can be used with Series and SeriesCoefficient.

Solutions to linear differential equations are represented as

DifferentialRoot objects:

In[17]:= FullSimplify[

SeriesCoefficient[

DifferentialRoot[Function[{y, x}, {y'[x] == y[x], y[0] == 1}

]][x], {x, 0, n}],

Assumptions :> Element[n, Integers] && n >= 0]

Out[17]= 1/n!

Solutions to algebraic equations are represented as Root objects:

In[18]:= Series[Root[#^5+a #+1&,1],{a,0,5}]

Out[18]= -1+a/5+a^2/25+a^3/125-(21 a^5)/15625+O[a]^6

Genfunlib overrides the SeriesCoefficient command for extracting the

coefficients of a power series.

Using Mathematica’s built–in SeriesCoefficient we can compute the co-

efficients of explicit power series like exp(z).

In[19]:= $Assumptions=(n>=0);

In[20]:= SeriesCoefficient[Exp[z],{z,0,n}]

Out[20]= 1/n!

Genfunlib adds the ability to work with symbolic power series, as in the

following example.

In[21]:= $FullAnalytic=True;

In[22]:= SeriesCoefficient[f[z]*g[z],{z,0,n}]

Out[22]:= Sum[SeriesCoefficient[f[z], {z, 0, $5}, Assumptions ->

n >= 0 && Element[$5, Integers] && $5 >= 0]*

SeriesCoefficient[g[z], {z, 0, n - $5},

Assumptions -> n >= 0 && Element[$5, Integers] && $5 >= 0],

{$5, 0, n}]

In[23]:= Replace[%,Verbatim[Rule][Assumptions,_]:>

Sequence[],Infinity]

Out[23]= Sum[SeriesCoefficient[f[z], {z, 0, $5}]*

SeriesCoefficient[g[z], {z, 0, n - $5}], {$5, 0, n}]

In this snippet, we expand [zn] (f(z)g(z)). We first set the global option

$FullAnalytic=True. The $FullAnalytic option, introduced in Genfunlib,

determines whether or not expressions such as f[z] should be assumed to be

power series. In general, they could be Puiseux series

∑
n≥n0

cnz
n/k, k ∈ Z>0, n0 ∈ Z,

with negative and/or fractional exponents such as zˆ(-1/2). Assuming that we

are working with power series (or analytic functions) allows us to unconditionally

use the Cauchy product formula, for example.

Mathematica note: The Mathematica command Series has the option

Analytic which is used for the same purpose as $FullAnalytic.

Next, we extract coefficients from f[z]*g[z] which returns a large expres-

sion containing Assumptions expressions. Using Replace to remove the as-

sumptions, we get a result corresponding to
∑n
j=0([zj ]f(z))([zn−j ]g(z)).

As well as products of symbolic power series, Genfunlib also adds support

for sums, shifts, derivatives, integrals, and powers. The general form of the ex-

pansion of [zn]f(g(z)) is an expression with multiple iterated sums that cannot

be nicely represented in Mathematica notation, but special cases of power series

composition such as f(k z), for constant k, are handled by Genfunlib. Because

Genfunlib enhances SeriesCoefficient with the ability to extract the coeffi-

cients of symbolic power series, we could use SeriesCoefficient on both sides

of a power series equation and obtain a recurrence relation.

Mathematica’s GeneratingFunction command is used for converting in the

other direction: from coefficients to power series. In this case, the built–in ver-

sion of GeneratingFunction comes with the ability to handle symbolic input.

In[24]:= GeneratingFunction[Sum[f[j]*g[n-j],{j,0,n}],n,z]

Out[24]= GeneratingFunction[f[n],n,z] GeneratingFunction[g[n],n,z]

In this example, Mathematica’s GeneratingFunction recognizes that the co-

efficient Sum[f[j]*g[n-j],{j,0,n}] is in the form of a Cauchy product, and

thus is the coefficient of the product of the generating functions of f[n] and

g[n].

Genfunlib adds functionality to GeneratingFunction in a few areas. First,

it handles coefficients of the form

p(n)

q(n)
f(n),

where p(n) and q(n) are polynomials in the index symbol n, and f(n) is symbolic.

The generating function
∑
n≥0

p(n)
q(n)f(n)zn can be expressed in terms of integrals

and derivatives of
∑
n≥0 f(n)zn. Problems arise if q(n) has positive integer

roots, so we ignore that case. (Specifically, some coefficients would be undefined

so it is not clear what the generating function should be.)

Mathematica note: Unfortunately, Mathematica 9 prevents the common

case of q(n) = 1 from working as expected. The GeneratingFunctions package

of Mallinger [21] can be used for this, however.

In[25]:= GeneratingFunction[f[n]/(n+1),n,z]

Out[25]= Integrate[GeneratingFunction[f[n], n, $3], {$3, 0, z}]/z

This example shows that ∑
n≥0

1

n+ 1
f(n)zn

is automatically computed by Genfunlib to be

z−1
∫ z

0

∑
n≥0

f(n)sn

 ds.

Genfunlib also implements the “multi–section” formula for summing only on

indices with a given residue modulo an integer.

In[26]:= GeneratingFunction[Boole[Divisible[n,2]]f[n],n,z]

Out[26]= 1/2 (GeneratingFunction[f[n],n,-z]+

GeneratingFunction[f[n],n,z])

From the point of view of Genfunlib’s source code, adding functionality to

GeneratingFunction and SeriesCoefficient is straightforward using pattern

matching and replacement rules. For example, the snippet

SeriesCoefficient[expr1_ + expr2_, {x_, 0, n_},

opts:OptionsPattern[]] := SeriesCoefficient[expr1, {x, 0, n},

opts] + SeriesCoefficient[expr2, {x, 0, n}, opts];

from Genfunlib (adapted slightly for simplicity) says that expressions that look

like

SeriesCoefficient[expr1 + expr2, {x, 0, n}]

are to be replaced by

SeriesCoefficient[expr1, {x, 0, n}] + SeriesCoefficient[

expr2, {x, 0, n}]

4.2 Numerical coefficients

Given a power series equation, it is a problem of computer algebra to evaluate

the coefficients of the solutions. Obviously, computation speed is relevant, and

for various sequence classes, there is active research around the problem of

efficiently computing sequences in the class. [2, 3, 5, 17] One must note that

there is a difference between computing all of the first n terms of a sequence,

and computing just the nth. Computing 2n, for some integer n, is quickly done

by using the method of repeated squaring. However, to compute 20, 21, . . . , 2n,

it is faster to simply perform one multiplication for each term. Genfunlib focuses

on the problem of computing all initial terms.

Mathematica note:

One special type of power series equation is the inverse equation, which has

the form

z = φ(f(z)),

where f(z) is the unknown power series. Various algorithms can be used to

compute the expansion of f(z), including the Lagrange inversion formula and

a more recent method due to Dominici [7]. Mathematica includes the func-

tion InverseSeries which can be used to solve this problem. In the following

example the equation is z = sin(f(z)).

In[27]:= Series[Sin[f], {f, 0, 10}]

Out[27]= f - f^3/6 + f^5/120 - f^7/5040 + f^9/362880 + O[f]^11

In[28]:= InverseSeries[%, z]

Out[28]= z + z^3/6 + (3*z^5)/40 + (5*z^7)/112 + (35*z^9)/1152 +

O[z]^11

A very general method for computing coefficients of power series uses the

Taylor coefficient formula

[zn]f(z) =
f (n)(0)

n!
.

If an equation F (f(z), z) = 0 is given in a power series f(z), the value [z0]f(z) =

f(0) can be computed by setting z = 0 and solving F (f(0), 0) = 0 for f(0). Then,

for n > 0, assuming we know [z0]f(z), [z1]f(z), . . . , [zn−1]f(z), we compute and

solve
dn

dzn
F (f(z), z)

∣∣∣∣
z=0

= 0

to find [zn]f(z). This can be generalized to differential equations, integral equa-

tions, etc.

Genfunlib implements this algorithm as the CoefsByDerivs command.

In[29]:= ?CoefsByDerivs

CoefsByDerivs[{eqn1, eqn2,...}, {f,g, ...}, {x, 0, nx},

{y, 0, ny}, ...] gives the Taylor expansions at (0,0,...)

of the power series f,g,... satisfying the equations.

In[30]:= CoefsByDerivs[{z*f[z]^2 + 1 - f[z] == 0}, {f[z]},

{z, 0, 10}]

Out[30]= {f[z] == 1 + z + 2*z^2 + 5*z^3 + 14*z^4 + 42*z^5 +

132*z^6 + 429*z^7 + 1430*z^8 + 4862*z^9 + 16796*z^10

+ O[z]^11}

In this example, we pass a system of equations consisting of just one equation,

zf(z)2 + 1− f(z) = 0,

then a list of power series which we want to expand, and a triple {z, 0, 10}

indicating that the power series is in the indetermiate z, the series is to be ex-

panded around 0, and should have order 10. Genfunlib only supports expansions

around 0, but multivariate problems are possible.

Mathematica note: Genfunlib calls Series to take the derivatives and

Solve to solve the equations.

The next example shows the expansion of f(u, v) and g(v) given by

f(u, v) = ug(v)

g(2v) =
1

2
exp(v),

up to [u4v3]f(u, v) and [v3]g(v).

In[31]:= CoefsByDerivs[{f[u, v] == u*g[v],

g[2 v] == 1/2*Exp[v]}, {f[u, v], g[v]}, {u, 0, 4}, {v, 0, 3}]

Out[31]= {f[u, v] == O[v]^4 + (1/2 + v/4 + v^2/16 + v^3/96)*u +

O[v]^4*u^2 + O[v]^4*u^3 + O[v]^4*u^4 + O[u]^5,

g[v] == 1/2 + v/4 + v^2/16 + v^3/96}

Newton iteration is a classical algorithm for numerically finding a zero of a

continuously differentiable real function f. Starting at an arbitrary real number

x0, Newton iteration involves computing the sequence defined by

xn+1 = xn −
f(xn)

f ′(xn)
.

Using a Taylor–Lagrange expansion of f around x∗, we get

0 = f(x∗) = f(xn) + f ′(xn)(x∗ − xn) +
1

2
f ′′(ξn)(x∗ − xn)2,

where ξn is between xn and x∗. This equation implies

x∗ − xn+1 =
−f ′′(ξn)

2f ′(xn)
(x∗ − xn)2. (1)

Thus, under some conditions on f ′ and f ′′, we have that the sequence xn in

Newton iteration converges to x∗ quadratically.

If f is not a real function but a mapping Q[[x]]→ Q[[x]] defined by y(x) 7→

F (x, y(x)), where F (x, y) ∈ Q[[x, y]], Newton iteration can be applied to com-

pute initial terms of the solution y(x)∗ of the power series equation

F (x, y∗(x)) = 0,

provided
∂

∂y∗(x)
F (x, y∗(x))

∣∣∣∣
x=y∗(x)=0

6= 0.

If we set y0(x) = [x0]y∗(x), the key observation to make is that instead of

Equation (1) referring to the distance between real numbers, we have

y(x)∗ − yn+1(x) = cn(x) (y(x)∗ − yn(x))
2
,

19



which implies that if y(x)∗ − yn(x) ≡ 0 (mod x2
n

), then y(x)∗ − yn+1(x) ≡ 0

(mod x2
n+1

). In other words, the nth iteration step generates 2n correct initial

coefficients [38].

Genfunlib implements the Newton iteration method for computing expan-

sions of solutions to power series equations with the CoefsByNewton command.

In[32]:= ?CoefsByNewton

CoefsByNewton[eqn, f, {x, 0, nx}] gives the Maclaurin

expansion in x of f satisfying eqn.

The code for CoefsByNewton comes partly from the implementation by

Koepf [18] (which contains a slight bug on line 4 – compare with Genfunlib’s

source code for details). The expression

CoefsByNewton[lhs==rhs, f[x,y,...], {x,0,n}]

where lhs == rhs is an equation in f[x,y,...], evaluates to the series expan-

sion of f[x,y,...] around 0 in x of order n. The following example shows the

expansion of f(z) given zf(z)2 + 1− f(z) = 0.

In[33]:= CoefsByNewton[z*f[z]^2 + 1 - f[z] == 0, f[z], {z, 0, 5}]

Out[33]= 1 + z + 2*z^2 + 5*z^3 + 14*z^4 + 42*z^5 + O[z]^6

Genfunlib’s CoefsByNewton cannot handle systems of equations, however if

a system of equations can be broken up into independent equations by other

methods, such as algebraic elimination [6], CoefsByNewton can be used.

4.3 Regular languages

A right regular grammar is a formal grammar (N,Σ, P, S), where N is a set

of non–terminal symbols, Σ is an alphabet, P is a set of production rules, and

S ∈ N is the start symbol, such that all the production rules in P are of one of

the following forms:

• B → a, where B is a non–terminal symbol in N and a is a terminal symbol

in Σ,

• B → aC, where B and C are in N and a is in Σ,

• B → ε, where B is in N and ε is the empty string.

A word w is generated by a right regular grammar if and only if, starting

from S we can obtain w by applying a finite sequence of the production rules.

A regular language is a set of words generated by a right regular grammar, or

satisfying any of a number of alternative specifications. These include deter-

ministic finite automata (DFA), nondeterministic finite automata (NFA), and

regular expressions [35]. Any one of these specifications can be called a regular

language representation (RLR).

Let L be a regular language, let |w| be the length of a word w in L, and let

cL(n) = |{w ∈ L : |w| = n}| .

In enumerative combinatorics, a key property of regular languages is that the

ordinary generating function of cL(n), that is
∑
n≥0 cL(n)zn, is a rational power

series. This fact is hinted at by the linear structure of right regular grammars.

Indeed, this rational power series is computable in closed form, given an

instance of an RLR. Let r be a regular expression, and let L(r) be the language

it successfully parses. We say that r is ambiguous iff there exists a word w ∈ L(r)

which can be parsed by r in more than one way. Examples of ambiguous regular

expressions include

ε∗, a∗∗, b|b, a∗|b∗,

while examples of unambiguous regular expressions include

a∗, a|aa|aaa, a|ab|ε.

It can be seen that if r is an unambiguous regular expression, applying the

following transformations to r yields its generating function [11]:

• concatenation becomes multiplication,

• union becomes addition, and

• Kleene star becomes the quasi–inverse operation 1/(1− ·).

Any RLR instance can be converted to a regular expression [35], so we can count

the words in a regular language if we have an algorithm that takes a regular

expression and returns an equivalent unambiguous one. It turns out that this

can be done by simply converting from regular expression to NFA to DFA and

back to regular expression according to the algorithms in [35]. Essentially, the

“determinism” of a DFA carries over to a regular expression in the form of

unambiguity. We note that Brabrand and Thomsen [4] give a useful algorithm

for finding the particular subexpressions of a regular expression that gives rise

to ambiguity.

Genfunlib implements a number of operations on RLR instances correspond-

ing to some of their abundant decidable properties, and in particular, Genfunlib

allows regular expressions to be disambiguated. Genfunlib allows regular lan-

guages to be represented by any of the following RLRs: NFA, DFA, regular

expression, right regular grammar, or directed graph with labeled vertices. Di-

rected graphs in which the vertices are labeled with letters are a nontraditional

RLR which have the benefit that they may be visualized using Mathematica’s

built–in graph plotting. Such a graph G = (V,E, l,Σ, S, E, e) is a set of vertices

V , a set of edges E, a labeling function l : V → Σ, where Σ is an alphabet, a

set S ⊆ V of starting vertices, a set E ⊆ V of ending vertices, and an integer e

which is 1 if ε is accepted and 0 otherwise. We say a labeled digraph G accepts

a nontrivial word w if and only if there exists a path v1, v2, . . . , vk ∈ V in G

with v1 ∈ S, and vk ∈ E, such that l(v1)l(v2) · · · l(vk) = w.

Genfunlib allows an instance any RLR to be converted to an instance of any

other. The following operations on RLRs are supported: union, intersection,

complement, reverse, concatenation, and Kleene star. Generating functions for

regular expressions can be computed by specifying a weight/marker for each

letter in the alphabet.

Mathematica note: Since Mathematica does not use a type system,

expressions must be inspected dynamically in order to determine whether they

satisfy certain properties. The author of a Mathematica function must decide

how to handle input expressions that the function is not supposed to take.

One example of this is a function on the integers being passed a symbol rather

than an integer — there are many reasons why a user might want to do this.

According to the Guidebook for programming [37],

As a rule of thumb, messages are not generated for “symbolic” input

if the function they appear in is used in classical mathematics.

Thus Table[1, {n}] generates a message, while Sin[z] does not. Genfunlib

attempts to follow this convention, so all functions related to regular languages

do not accept symbolic input or input that cannot be directly used for compu-

tation.

In Mathematica, input validation checks can be complete or shallow. A

shallow check would only make sure the input is in roughly the right form, but

would not check whether the input would cause an error later on. Generally,

openly available Mathematica packages do not perform as much validation as

functions in the Mathematica kernel.

The FiniteFields package largely does not do input validation. It sometimes

performs shallow syntactic validity checks, sometimes performs total semantic

validity checks, and sometimes sends error messages. (On failure, all checks

result in an expression returning unevaluated). The Splines package does only

shallow syntactic input validation. Mathematica built-in functions often vali-

date any sequence of arguments and send messages on errors, and generally do

not produce results that will cause errors “later on”.

Since Mathematica expressions must be inspected dynamically, one may seek

a method of caching properties of expressions explicitly to avoid overcomputa-

tion. For example, a function may always return a prime number, so the values

of this function do not need to be tested later on for primality. In Genfunlib,

the result of a successful RegUnion command, for example, is guaranteed to be

a valid RLR, but when it is passed to another function, it is checked for va-

lidity anyway. One possible strategy is for all functions to store the validity of

their results right before they return them, by setting a value of the "validate"

function, which would look something like this:

validate[ret] = True; Return[ret]

This system could be altered by making the validate symbol only remember the

last n such expressions to save memory.

If a public function calls another public function, it presumably always passes

valid input. One way to avoid unnecessary computation of the validity is to pass

an option saying “validation not required”; another is for public functions never

to call public functions.

The validation scheme of Genfunlib is as follows. Public functions

call validation directly (right in their definition) unless told not to by the

validationRequired option; private functions do not do validation. Using

validationRequired saves some computation at the expense of more compli-

cated code. Data representation symbols like DFA do not do validation them-

selves; indeed they are not defined as functions. This follows the convention

exhibited in Mathematica symbols like RegularExpression and Graph.

The authors of the Combinatorica package for Mathematica [25] say, “Our

aim in introducting permutation groups into Combinatorica is primarily for

solving combinatorial enumeration problems. We make no attempt to efficiently

represent permutation groups or to solve many of the standard computational

problems in group theory.” The situation for this package and automata/gram-

mar algorithms is similar – performance is not the highest priority.

In Genfunlib, letters are represented by nonempty Strings and words are

represented by Lists of letters.

Mathematica note: An alternative approach to providing regular lan-

guage functionality in Mathematica would have been to make a J/Link inter-

face to the dk.brics.automaton Java package available at http://www.brics.

dk/automaton/.

The expression formats used by Genfunlib for each RLR are as follows.

An NFA expression is of the form

NFA[numStates, alphabet, transitionMatrix,

acceptStates, initialState]

The number of states is a nonnegative integer; any NFA with zero states accepts

no words. The alphabet is a sorted list of distinct strings, not containing "". A

value of {} (the empty list) means the NFA accepts either the empty language

or {ε}. The transition matrix is a numStates by Length[alphabet]+1 matrix

where entry i, j is a list of integers corresponding to states accessible from state i

and, if j <= Length[alphabet], on the letter alphabet[[j]], otherwise, on an

ε–move. If the number of states is zero, then we must have transitionMatrix

== {}. If the alphabet is empty, then transitionMatrix has one column (if it

has any rows). The accept states are a list of integers between 1 and the number

of states. The initial state is an integer between 1 and number of states, or Null

iff numStates == 0.

As an example, we can represent the NFA in Figure 1 by the following

expression.

NFA[6, {"0", "1"}, {

{{2}, {2}, {3}},

{{4}, {2}, {}},

{{4}, {2}, {}},

{{6}, {3}, {}},

{{}, {}, {}},

{{4}, {5}, {}}

}, {3, 4}, 1]

A regular expression that is equivalent to the NFA in Figure 1 is

(()|1(1)∗0|1(1)∗00(00)∗0|0(0)∗|(1(1)∗01|1(1)∗00(00)

∗01|0(0)∗1)(((0|1(1)∗0)1|(0|1(1)∗0)0(00)∗01))∗(()|0

|1(1)∗0|(0|1(1)∗0)0(00)∗0)).

A DFA expression is of the form

DFA[numStates, alphabet, transitionMatrix,

acceptStates, initialState]

Figure 1: A six–state NFA.

The number of states is a nonnegative integer; any DFA with zero states accepts

no words. The alphabet is a sorted list of distinct strings, not containing "".

A value of {} means the DFA accepts either the empty language or {ε}. The

transition matrix is a numStates by Length[alphabet] matrix where entry i, j

is the state accessible from state i and letter alphabet[[j]]. If the number of

states is zero, then transitionMatrix == {}. If the alphabet is empty, then

transitionMatrix == {{}, {}, ...}. The accept states are a list of integers

between 1 and the number of states. The initial state is an integer between 1

and the number of states, or Null if numStates == 0.

As an example, we can represent the DFA shown in Figure 2 with the fol-

lowing expression.

DFA[5, {"0", "1"}, {

{2, 3},

{2, 4},

{4, 3},

{5, 5},

{5, 5}

}, {4}, 1]

Figure 2: A five–state DFA.

A regular expression that is equivalent to the DFA in Figure 2 is

(0(0)∗1|1(1)∗0).

In Genfunlib, a regular expression is represented by an expression with

head Regex, and an argument built up from nonempty strings, EmptyWord,

RegexStar, RegexConcat, and RegexOr. The expression Regex[Null] matches

no words (the empty language).

As an example, the regular expression (a|bb)∗bbb is represented as

Regex[RegexConcat[RegexStar[RegexOr["a", RegexConcat["b", "b"]]],

"b", "b", "b"]]

Applying Mathematica’s TreeForm command to this expression produces

Figure 3.

Right regular grammars are represented in Genfunlib by expressions with

head RRGrammar, and one argument which is a list of rules in the form sym ->

RHS or sym[n] -> RHS, where sym is a symbol, n is an integer, and RHS is either

• EmptyWord,

Figure 3: The parse tree of (a|bb)∗bbb.

• a string,

• an expression on the left hand side of another rule,

• RRGrammarConcat[str, sym] (where str is a string),

• RRGrammarConcat[str, sym[n]], or

• RRGrammarOr[args], where args is a sequence of any of the things in this

list.

Strings cannot be empty. An empty list corresponds to the empty language.

For example, let G be the right regular grammar such that N = {S,A},

Σ = {a, b, c}, and the production rules P are

S → aS

S → bA

A→ ε

A→ cA.

It can be seen that G is equivalent to the regular expression a∗bc∗. In Gen-

funlib, G becomes

RRGrammar[{

s -> RRGrammarConcat["a", s],

s -> RRGrammarConcat["b", a],

a -> EmptyWord,

a -> RRGrammarConcat["c", a]

}]

A directed graph with labeled vertices is represented by an expression of

the form Digraph[graph, startVertices, endVertices, eAccepted]. The

graph is a directed Mathematica Graph, whose vertices are all labeled with

nonempty strings. The starting vertices are a list of vertices of the graph; if

this list is empty, the language associated with the graph contains no words

except possibly ε. The ending vertices are a list of vertices of the graph; if

this list is empty, the language associated with the graph contains no words ex-

cept possibly ε. Finally, eAccepted is True if ε is accepted and False otherwise.

Any of the five RLRs can be converted to any other using the commands

ToNFA, ToDFA, ToRegex, ToRRGrammar, and ToDigraph.

In[34]:= ?ToDFA

ToDFA[NFA[...]] is a DFA defined from a NFA.

ToDFA[Regex[...]] is a DFA defined from a symbolic regular

expression.

ToDFA[RRGrammar[...]] is a DFA defined from a right

regular grammar.

ToDFA[Digraph[...]] is a DFA defined from a digraph

with labeled vertices.

In[35]:= ToDFA[Regex[RegexOr["a", RegexStar["b"]]]]

Out[35]= DFA[4, {"a", "b"}, {{2, 3}, {4, 4}, {4, 3}, {4, 4}}, {1, 2,

3}, 1]

The algorithms for performing these conversions come from the textbook

of Sipser [35] with two exceptions. First, given an NFA M with state set Q,

the standard algorithm for obtaining a DFA that is equivalent to M involves

computing the powerset 2Q, which is used as the state set for the DFA. In some

cases, there is no DFA equivalent toM with a state set with fewer elements than

2Q [34]. However, the algorithm used in Genfunlib produces an automaton with

many fewer states, when this is possible. The first step is to remove states from

the NFA that are inaccessible from the start state or from every accepting state.

Each state removed halves the maximum size of the resulting DFA. Next, when

computing the set of states for the DFA, Genfunlib only includes those sets in

2Q that are accessible from the initial state by performing an incremental search

starting at the initial state. Finally, the DFA is minimized using Hopcroft’s al-

gorithm [35]. Experimental tests showed a huge speed increase when using these

steps to avoid creating a DFA with 2|Q| states. In testing, Genfunlib found the

minimal DFA of an NFA with 379 states in under half an hour on a computer

with an AMD Phenom II X4 995 CPU and 8 gigabytes of RAM. However, the

resulting DFA had only 8 states, so in terms of the length of the output, Gen-

funlib’s performance still has room for improvement. The dk.brics.automaton

Java package [22] uses a similar determinization algorithm, but the performance

of that library is much higher than that of Genfunlib, presumably because Java

is a lower–level language than Mathematica.

The other somewhat unusual conversion algorithm used in Genfunlib is the

one that maps labeled digraphs to automata and vice versa, since the digraph

RLR is not included in textbooks on automata theory. Because in Genfunlib’s

digraphs, vertices are labeled, and in the graphs derived from automata, edges

are labeled, something like the line digraph construction can be used. The line

digraph of a digraph G has one vertex for each edge of G, and has an edge

between two of its vertices if and only if the corresponding edges in G form a

length–two directed path in G. “Graphs” derived from automata are actually

hypergraphs, but the line digraph definition can be adapted to this case. When

Genfunlib converts something to a DFA, the output is a minimal DFA, but

when converting to labeled digraphs or NFAs, the output may not be minimal.

Figure 4: A DFA with eight transitions.

b [2]start

b [6]

a [5] b [8]

a [7]

a [3]

b [4]

a [1]

startstart

Figure 5: The digraph constructed from the DFA in Figure 4. Vertices are
labeled by the transition they correspond to in the DFA.

Minimizing NFAs is a PSPACE-complete problem [14].

As an illustration, the DFA in Figure 4 has each of its “edges” shown with

an integer label, in brackets. Genfunlib converts this DFA to the digraph shown

in Figure 5. The vertices of this digraph are shown with extra labels in brackets

indicating the edge in the original DFA to which they correspond.

In addition to conversions between the five RLRs, Genfunlib includes func-

tions for converting between Genfunlib regular expressions and Mathematica

built–in regular expressions, which have head RegularExpression. To be con-

verted, the Mathematica regular expressions must have a restricted syntax:

Only letter characters, parentheses, asterisks, and bars (|) are allowed.

The following example converts the regular expression (c|a)∗ from Mathe-

matica syntax to Genfunlib syntax and back, and demonstrates how built–in

Mathematica functions such as StringMatchQ can be used with Mathematica

regular expressions.

In[36]:= ToRegex[RegularExpression["(c|a)*"]]

Out[36]= Regex[RegexStar[RegexOr["c", "a"]]]

In[37]:= ToRegularExpression[%]

Out[37]= RegularExpression["((c|a))*"]

In[38]:= StringMatchQ["accaccccaaa", %]

Out[38]= True

Since, in Genfunlib, letters are represented by strings which may be longer

than one character (and words are represented by lists of strings), some issues

may arise when converting to Mathematica regular expressions. In Genfunlib,

{"a","aa"} and {"aa","a"} are different words. However, since in Mathemat-

ica represents letters as length–one strings and represented words as strings,

Regex[RegexOr[RegexConcat["a", "aa"], RegexConcat["aa", "a"]]]

gets converted to

RegularExpression["(aaa|aaa)"]

Regular languages have a rich set of closure properties [35], which Genfun-

lib implements in the form of six operations: Kleene star (RegStar), comple-

ment (RegComplement), reverse (RegReverse), union (RegUnion), concatena-

tion (RegConcat), and intersection (RegIntersection).

The algorithms for performing these operations all come from Sipser’s text-

book [35]. For each operation, Genfunlib implements such an algorithm that

works for one type of RLR. Then the operation is implemented for all other RLR

types by converting them to the type for which an algorithm is implemented,

running the algorithm, then converting back again. As an example, the con-

catenation operation is directly implemented only for right regular grammars,

but users can call RegConcat on two DFAs because they will get automatically

converted to and from grammars.

The unary operations RegStar and RegReverse simply take one RLR as an

argument. The operation RegComplement takes one RLR r and an alphabet Σ

and returns an RLR r′ of the same type as r such that L(r′) = Σ∗ \L(r), where

L(r) is the language of r.

In[39]:= ?RegComplement

RegComplement[NFA[...], {"str1", "str2",...}] performs the

complement operation on an NFA with respect to a given

alphabet.

RegComplement[DFA[...], {"str1", "str2",...}] performs the

complement operation on a DFA with respect to a given

alphabet.

RegComplement[Regex[...], {"str1", "str2",...}] performs the

complement operation on a symbolic regular expression

with respect to a given alphabet.

RegComplement[RRGrammar[...], {"str1", "str2",...}] performs the

complement operation on a right regular grammar with respect

to a given alphabet.

RegComplement[Digraph[...], {"str1", "str2",...}] performs the

complement operation on a directed graph with respect to a

given alphabet.

In[40]:= RegComplement[Regex[RegexStar["a"]], {"a", "b"}]

Out[40]= Regex[ RegexConcat[RegexStar["a"], "b",

RegexStar[RegexOr["a", "b"]]]]

The binary operations RegUnion, RegConcat, and RegIntersection take two

RLRs of the same type.

Genfunlib overloads GeneratingFunction to take a regular expression and a

list of rules mapping letters to weights. The regular expression is automatically

disambiguated, and the letters are replaced with their weights to obtain a ratio-

nal power series. Here we compute the generating function for the ambiguous

regular expression a|a|b|b∗ that marks a with u and b with v:

In[41]:= GeneratingFunction[Regex[RegexOr["a","a","b",

RegexStar["b"]]], {"a"->u,"b"->v}]

Out[41]= 1+u+v/(1-v)

4.4 Symbolic method

Genfunlib contains a basic implementation of the symbolic method. An ap-

propriate introduction to the symbolic method may be found in the author’s

honours project [20] or Flajolet and Sedgewick’s book [11].

Genfunlib allows combinatorial classes to be defined by specifications, and

can convert these specifications to corresponding systems of equations in gener-

ating functions.

A specification is represented by an expression of the form

Labeled Unlabeled
Sum SMPlus Sum SMPlus
Product SMTimes Product SMTimes
Sequence SMSeq Sequence SMSeq
Cycle SMCyc Cycle SMCyc
Set SMSet Powerset SMSet
Pointing SMPointing Multiset SMMultiset
Substitution SMSub Pointing SMPointing

Substitution SMSub

Table 1: Admissible constructions supported by Genfunlib.

Spec[{lhs1==rhs1,...},labeled?]

In the list of equations, the left hand sides are symbols representing classes,

and the right hand sides are expressions built with combinatorial constructions,

symbols found in a left hand side, restrictions (Restricted[...]), and atomic

and neutral classes. The second argument is True if the classes are labeled, and

False if they are unlabeled.

Genfunlib supports multiple atomic classes, each of which is indexed by a

positive integer and is marked seperately in generating functions. They have

the form ZClass[1], ZClass[2],.... The neutral class is represented by the

symbol EClass.

The combinatorial constructions in Genfunlib are shown in Table 1, along

with their syntax. Sum and product are commutative binary constructions; se-

quence, cycle, set, multiset, and pointing are unary constructions. Substitution

is a binary construction, but it is not commutative. The expression SMSub[A,B]

corresponds to A(B), if A is represented by the symbol A and B is represented

by the symbol B.

The set, multiset, sequence, and cycle constructions take an optional second

argument in the form Cardinality->f, where f is a predicate on the integers

which gives the allowed number of components. For example,

SMSeq[class, Cardinality->(#<=k&)]

represents the class of finite sequences of elements of class with at most k

terms. It is fine to use a symbol like k rather than an integer.

The other kind of restriction that can be done is with the Restricted sym-

bol. The expression Restricted[class,f], where f is a predicate on the inte-

gers, represents the class of objects from class with sizes that f maps to True.

For example,

Restricted[SMSeq[SMTimes[ZClass[1],ZClass[2]]], #>2&]

represents the class containing all the objects in Seq(Z1 × Z2) except the sin-

gleton sequence.

Mathematica note: Predicates on the integers must return unevaluated

when given symbolic input. The predicate Mod[#, 2]& satisfies this require-

ment, while PrimeQ[#]& does not.

The command for passing from a combinatorial specification to a system of

generating function equations is ToGFEqns. It takes two arguments, the first is a

Spec expression and the second is a symbol to be used as the indeterminate. If

the symbol x is supplied, ZClass[1] is marked with x[1], ZClass[2] is marked

with x[2], and so on. In the output, the generating function of a class is written

as the symbol for the class, with arguments.

In[42]:= ?ToGFEqns

ToGFEqns[spec, indeterminate] gives the system of power series

equations corresponding to spec, using the given indeterminate.

In[43]:= ToGFEqns[

Spec[{c == SMTimes[ZClass[1], SMSeq[ZClass[1]]]}, True], z]

Out[43]= {c[z[1]] == z[1]/(1 - z[1])}

The code that performs the translation to generating functions uses pattern

matching and replacement rules. The following example snippet is a simplified

segment of Genfunlib’s code that takes a specification called spec and replaces,

e.g. ZClass[2] and ZClass[3] with z[2] and z[3], and replaces EClass with

1.

Replace[spec,

{

ZClass[n_] :> z[n],

EClass :> 1,

},

{1, Infinity}, Heads -> True

];

Genfunlib includes one more command related to the symbolic method:

ToGenfunlibSpec. This command takes two arguments. The first is a string

containing the Maple code for a Combstruct grammar in the format used on

the Encyclopedia of Combinatorial Structures (ECS) website at

http://algo.inria.fr/ecs/

The second argument is a Boolean value indicating whether the classes are

labeled. The output is a Spec expression representing the same specification as

the Combstruct grammar. Since Genfunlib and Combstruct specifications are

similar in structure, the conversion is not complex and mostly consists of term–

for–term translation. Class symbols are converted from upper case to lower case

(and reduplicated) to conform to Mathematica conventions.

As an example, we can take entry #9 from the ECS: words on the alphabet

{a, b} with at most one consecutive b-letters. The ECS gives the following

unlabeled Combstruct specification:

{S = Prod(X,Sequence(Prod(a,X))), X = Sequence(b,card <= 1),

a = Atom, b = Atom }

In Genfunlib, the class S becomes ss.

In[44]:= ?ToGenfunlibSpec

ToGenfunlibSpec[mapleString, labeled] converts from Combstruct to

Genfunlib combinatorial specification syntax.

In[45]:= ToGenfunlibSpec["{S = Prod(X,Sequence(Prod(a,X))), X =

Sequence(b,card <= 1), a = Atom, b = Atom }", False]

Out[45]= Spec[{ss == SMTimes[xx, SMSeq[SMTimes[a, xx]]],

xx == SMSeq[b, Cardinality -> (#1 <= 1 &)], a == ZClass[1],

b == ZClass[1]}, False]

We can convert to generating functions using indeterminate z and solve for

ss[z[1]].

In[46]:= ToGFEqns[%, z]

Out[46]= {ss[z[1]] == xx[z[1]]/(1 - a[z[1]] xx[z[1]]),

xx[z[1]] == 1 + b[z[1]], a[z[1]] == z[1], b[z[1]] == z[1]}

In[47]:= Solve[Eliminate[%, {a[z[1]], b[z[1]], xx[z[1]]}],

ss[z[1]]]

Out[47]= {{ss[z[1]] -> (-1 - z[1])/(-1 + z[1] + z[1]^2)}}

There is room for expansion of Genfunlib’s symbolic method capabilities.

Symbolic method specifications are useful because tractable operations can be

performed on them. This is of note since in general, combinatorial objects are

too complex for such simple notation. Assuming the user is able to provide a

combinatorial specification, one valuable question may be whether the specifi-

cation is well–formed. The class I defined by I = Seq(E) has an infinite set of

objects of size 0, so the specification should be edited before anything is done

with it. Work on checking specifications for well–definedness has been published

by Salvy and coauthors [10, 26]. Genfunlib does not implement these complete

validity checks on specifications; however, errors are likely to be found at a later

state when working with power series. For example, I = Seq(E) would generate

a division by 0 error.

Lastly, there exists a notation system and framework for combinatorial ob-

jects similar to symbolic combinatorics called species [20]. Currently, Genfunlib

provides support for the symbolic method only, although the possibility of im-

plementing species instead and including a converter for specifications from the

symbolic method to species was considered. It does appear as though a nat-

ural mapping exists between the sequence, set, and cycle constructions on the

one hand, and the species of linear orders, sets, and cyclic permutations on the

other. However, problems arise eventually. Whereas the pointing construction

in the symbolic method corresponds, in unlabeled structures, to the power se-

ries operator xD, for a species F the type generating series F̃ ◦(x) cannot be

expressed in terms of F̃ (x), but is given by

F̃ ◦(x) = x

(
∂

∂x1
ZF

)
(x, x2, x3, . . . ).

Thus, in species, the pointing operation cannot correspond to any admissible

construction in the symbolic method. This kind of subtle difference also exists

between the substitution “construction” in the symbolic method and “operation”

in species. Salvy et al. discuss perhaps less serious issues with converting the

unlabeled multiset construction to the language of species [26].

4.5 Asymptotics

Flajolet et al. [10] define the class E of elementary functions as the set of func-

tions containing the monomials 1 and z, and closed under the operations of

{+,×, Q, L,E}, where

Q(f) =
1

1− f
, L(f) = log

1

1− f
, E(f) = exp(f).

They proceed to give algorithms that can be used to automatically extract

Maclaurin coefficient asymptotics for functions in a subset of E . Before going

further here, we note that at this point some confusion can appear related to

the terminology and notation used. First, even though when talking about

asymptotics we need to think of generating functions as complex functions, for

the purpose of computer algebra we really still compute with power series. That

is, “exp(z)” cannot, strictly speaking, be called a function. Also, we actually

want to talk about expressions built with the operations +,×, Q, L,E, etc. —

we are talking about syntax, not semantics. Deciding whether a given expression

semantically belongs in E is in general impossible, and not our main goal.

The class E is subdivided into three disjoint sets

E = EAL ∪̇ Eentire ∪̇ Eother,

where Eentire is the set of functions in E that are entire, and EAL is the functions

with algebraic–logarithmic growth, meaning that f ∈ EAL if and only if f has

radius of convergence 0 < r <∞ then there exist α ∈ R, k ∈ Z≥0 such that

f(z) ∼ 1

(1− z/r)α
logk

1

1− z/r
, as z → r−.

The algorithm for computing asymptotic expansions of the Maclaurin coeffi-

cients of functions in EAL described by Flajolet et al. [10] was implemented in

Combstruct for Maple and is now partly implemented in Genfunlib for Mathe-

matica as the CoefLimit command. This command takes an aperiodic function

f ∈ EAL and returns something asymptotically equivalent to the general term of

the Maclaurin coefficient sequence of f. By saying that f is aperiodic, we mean

that f(z) cannot be expressed as zag(zb) where b > 1 and a are integers, and

g ∈ EAL.

In[48]:= ?CoefLimit

CoefLimit[f,z,n] gives an expression that is asymptotically

equivalent to SeriesCoefficient[f, {z,0,n}] as n goes to

infinity.

In[49]:= CoefLimit[1/(1 - z*Exp[z]), z, n]

Out[49]= ProductLog[1]^(-1 - n)/(1 + E^ProductLog[1])

In[50]:= CoefLimit[1/(1 - Log[1/(1 - z*Exp[z])]), z, n]

Out[50]= -(((1 - (-1 + E)/E) ProductLog[(-1 + E)/E]^(-1 -

n))/(-((-1 + E)/E) - E^ProductLog[(-1 + E)/E]))

In the case where f ∈ EAL has positive radius of convergence r and

f(z) ∼ 1

(1− z/r)α
logk

1

1− z/r
as z → r−,

where α ≤ 0, CoefLimit merely returns the exact expression for the coefficients,

using Mathematica’s SeriesCoefficient.

In[51]:= CoefLimit[Log[1/(1 - z*Exp[z])], z, n]

Out[51]= SeriesCoefficient[Log[1/(1 - E^z z)], {z, 0, n}]

In[52]:= $Assumptions = n > 1;

In[53]:= CoefLimit[1/(1 - z) Log[1/(1 - 2 z)], z, n]

Out[53]= -I*Pi - 2^(1 + n) LerchPhi[2, 1, 1 + n]

Here LerchPhi[2, 1, 1 + n] represents
∑
k≥0 2k/(k + n+ 1).

Mathematica note: The original algorithm [10] works for periodic gen-

erating functions and can obtain an asymptotic expansions of the form

Cρ−nns logk n

(
1 +O

(
1

log n

))
.

However, in Mathematica asymptotic series are represented with SeriesData

objects, and these only support big–Oh terms of the form O(nk). Thus,

CoefLimit does not give big–Oh terms, merely the dominant term. Since peri-

odic generating functions have zeroes in their coefficients, the coefficients are not

asymptotically equivalent to anything, so Genfunlib cannot give an asymptotic

expression in this case.

A general way to specify a generating function is to define it implicitly by

a power series equation. Genfunlib provides the TreeAsymptot command for

computing the coefficient asymptotics of generating functions f(z) satisfying

equations of the form

f(z) = zφ(f(z)),

where φ ∈ EAL satisfies φ(z) 6= c0 + c1z and φ(0) 6= 0. In this case, we can

apply Theorem VI.6 [11] to derive the asymptotics of [zn]f(z) provided we can

compute the minimum solution τ to the characteristic equation

φ(τ)− τφ′(τ) = 0, τ > 0,

and the periodicity of φ. The value of τ can be computed by Mathematica’s

built–in equation solver, and we can obtain the necessary information about the

period of φ via the “Reduction” algorithm from Flajolet et. al [10].

The TreeAsymptot command uses this method to give an asymptotic ex-

pression consisting of a main estimate and a big O term.

In[54]:= ?TreeAsymptot

TreeAsymptot[y[z]==z*phi,n] gives an asymptotic series for

SeriesCoefficient[y[z], {z,0,n}] as n goes to infinity,

where y[z] is the solution to y[z]==z*phi that is

analytic at 0.

In[55]:= TreeAsymptot[f[z] == z*Exp[f[z]], n]

Out[55]= E^n ( (n^(-1))^(3/2)/Sqrt[2*Pi] + O[1/n]^(5/2))

The case where φ(z) is periodic is allowed.

In[56]:= TreeAsymptot[f[z] == z*Exp[f[z]^2], n]

Out[56]= (2 E)^(n/2) Boole[Mod[n, 2] == 1] ((n^(-1))^(3/2)/

Sqrt[2*Pi] + O[1/n]^(5/2))

Flajolet and Sedgewick [11] give an algorithm for obtaining coefficient asymp-

totics of generating functions defined by algebraic equations. This surely useful

algorithm has so far not been implemented in any computer algebra system.

5 Genfunlib in action

5.1 Mathematics contest problems

Elementary problems from combinatorics are often chosen for high school math

contests. This subsection contains a selection of contest–style problems and

shows how Genfunlib can help to solve them; more contest problems solved

with the aid of a computer algebra system can be found in an article by Dumas

and Salvy [8].

Problem 1. How many arrangements are there of the letters in the word

MATHEMATICS? [30]

Solution: As Zeilberger says [12], “Enumerating specific finite sets is no

longer considered mathematics. A genuine mathematical fact has to incorporate

infinitely many facts, and the generic enumeration problem is to enumerate not

just one set but all the sets in an infinite family.” In particular, this problem can

be solved easily “by computer” simply by performing a brute–force enumeration

(in a math contest, this method is too slow). However, Genfunlib could also be

used.

The word MATHEMATICS contains 11 letters. If we create sequences of 11

atoms 1 , we can associate the label of the ith atom with the position of the

ith letter of MATHEMATICS in an arrangement of the letters. For example,

(
8 , 4 , 5 , 2 , 7 , 1 , 3 , 10 , 9 , 11 , 6

)

corresponds to the arrangement MHAATSEMITC. The problem is that not all

letters in MATHEMATICS are distinct, so not every different labeled sequence

corresponds to a different arrangement. We fix this by creating sequences of sets

of atoms, where there is one set for each different letter in MATHEMATICS.

This way, each set of positions of a letter is counted, not each sequence of

positions.

Our specification of the classM of rearrangements of MATHEMATICS is

M = Set2

(
1
)3 × Set1

(
1
)5

= Set2

(
1
)3 × 1

5
.

With Genfunlib, we convert this to an exponential generating function m[z]

where the coefficient of z11 multiplied by 11! is the solution to the problem.

In[57]:= ToGFEqns[Spec[{

m == SMTimes[

SMSet[ZClass[1], Cardinality -> (# == 2 &)],

SMSet[ZClass[1], Cardinality -> (# == 2 &)],

SMSet[ZClass[1], Cardinality -> (# == 2 &)],

ZClass[1],

ZClass[1],

ZClass[1],

ZClass[1],

Figure 6: An NFA N for the allowed sequences of coin tosses in Problem 2.

ZClass[1]

]}, True], z] /. z[1] -> z

Out[57]= {m[z] == z^11/8}

From here we observe that the solution is

In[58]:= 11!/8

Out[58]= 4989600

Problem 2. A fair coin is to be tossed ten times. Find the probability that

heads never occur on three consecutive tosses. (This is a modified version of

AIME 1990 #9, a typical Fibonacci problem [31].)

Solution: Say the coin is tossed n times. Then we can represent the sequence

of tosses as an n–word over the alphabet {H,T}. Let N be the NFA in Figure

6. Then the language accepted by N is the set of words representing coin tosses

in which heads never occur on three consecutive tosses. Using Genfunlib, we

can convert N to an equivalent regular expression.

In[59]:= ToRegex[NFA[3,{"H","T"},{

{{2},{1},{}},

{{3},{1},{}},

{{},{3},{}}

},{1,2,3},1]]

Out[59]= Regex[RegexOr[RegexStar[T],RegexConcat[

RegexStar[T],H,RegexStar[RegexConcat[T,RegexStar[T],H]],

RegexOr[EmptyWord,RegexConcat[T,RegexStar[T]]]],

RegexConcat[RegexStar[T],H,RegexStar[RegexConcat[T,

RegexStar[T],H]],H,RegexStar[T]]]]

Now, we obtain the generating function of this regular expression to find the

number of words of length 10.

In[60]:= GeneratingFunction[%, {"H" -> z, "T" -> z}]

Out[60]= (1 + z + z^2/(1 - z))/(1 - z - z^2)

In[61]:= SeriesCoefficient[%, {z, 0, 10}]

Out[61]= 232

Thus, the answer is 232/210 = 29/128.

Problem 3. Find a recursion for the number of partitions of an n–set [9].

First we can note that the combinatorial class P of set partitions is defined

P = SMSet
(
SMSet≥1( 1 )

)
.

That is, a set partition is simply a set of nonempty sets. With Genfunlib we

can get the exponential generating function P (z) for the class P, represented

by p[z].

In[62]:= ToGFEqns[

Spec[{p == SMSet[SMSet[ZClass[1], Cardinality ->

(# >= 1 &)]]}, True], z] /. z[1] -> z

Out[62]= {p[z] == E^(-1 + E^z)}

This is actually an explicit equation for the generating function p[z]. To arrive

at a recurrence relation, we logarithmically differentiate both sides and solve for

p[z].

In[63]:= Solve[D[Log[p[z]], z] == D[Log[E^(-1 + E^z)], z], p[z]]

Out[63]= {{p[z] -> E^-z p'[z]}}

Using Genfunlib’s enhancement to SeriesCoefficient, we extract coeffi-

cients of the right hand side.

In[64]:= SeriesCoefficient[E^-z p'[z], {z, 0, n}]

Out[64]= Sum[((-1)^$6*(1 + n - $6)*SeriesCoefficient[p[z],

{z, 0, 1 + n - $6}, Assumptions -> Element[$6, Integers] &&

$6 >= 0])/$6!, {$6, 0, n}]

If we let pn = n![zn]P (z), we can use the above result to arrive at the relation

p0 = 1, pn = pn−1 −
n−1∑
j=1

(
n− 1

j

)
(−1)jpn−j .

Problem 4. Prove that the number of binary n–words with exactly m 01–

blocks is
(
n+1
2m+1

)
[9].

Solution: The way we attack this problem is defining the NFA Nb in Figure

7. The language that Nb accepts is all words over the alphabet {0, 1, b} where

Figure 7: An NFA Nb that marks 01 blocks with b.

the letter b only appears as part of the subword 0b1, and the subword 01 never

appears. Counting the number of words with n letters 0 and 1 andm occurrences

of b will solve the problem.

We define nfab to be Nb from Figure 7 with Genfunlib.

In[65]:= nfab = NFA[4, {"0", "1", "b"}, {

{{2}, {1}, {}, {}},

{{2}, {}, {3}, {}},

{{}, {4}, {}, {}},

{{2}, {1}, {}, {}}

}, {1, 2, 4}, 1];

We obtain a generating function from nfab, with u marking 0 and 1, and v

marking b.

In[66]:= GeneratingFunction[ToRegex[nfab],

{"0" -> u, "1" -> u, "b" -> v}]

Out[66]= (1 + u/(1 - u))/(1 - u - (u^2 v)/(1 - u))

Now, we extract the coefficient of vm.

In[67]:= $Assumptions = m >= 0;

In[68]:= SeriesCoefficient[1/(1 + u (-2 + u - u v)), {v, 0, m}]

Out[68]= (u^2/(-1 + u)^2)^(1 + m)/u^2

Lastly we compare this to
∑
n≥2m

(
n+1
2m+1

)
un.

In[69]:= Out[68] == Sum[Binomial[n + 1, 2 m + 1] u^n,

{n, 2 m, Infinity}] // FullSimplify

Out[69]= True

This may seem like an indirect route to the result. The explanation is that,

as long as we want to use Mathematica and Genfunlib as much as possible, the

most direct route sometimes does not work but something more creative does.

Problem 5. How many n–words from the alphabet {0, 1, 2} are such that the

neighbors differ by at most 1? [9]

Solution: One can see that this is a regular language accepted by the NFA

Nd shown in Figure 8. We define Nd in Genfunlib as nfadif.

nfadif = NFA[4, {"0", "1", "2"}, {

{{2}, {3}, {4}, {}},

{{2}, {3}, {}, {}},

{{2}, {3}, {4}, {}},

{{}, {3}, {4}, {}}

}, {1, 2, 3, 4}, 1];

Figure 8: The NFA Nd that accepts words from the alphabet {0, 1, 2} such that
neighboring letters differ by at most 1.

All that remains in order to count the n–words is to obtain the generating

function and extract coefficients.

In[70]:= GeneratingFunction[

ToRegex[nfadif], {"0" -> z, "1" -> z, "2" -> z}] //

FullSimplify

Out[70]= -((1 + z)/(-1 + z (2 + z)))

In[71]:= $Assumptions = n >= 1;

In[72]:= SeriesCoefficient[-((1 + z)/(-1 + z (2 + z))), {z, 0, n}]

Out[72]= (-2 (1 - Sqrt[2])^n + Sqrt[2] (1 - Sqrt[2])^n +

2 (1 + Sqrt[2])^n + Sqrt[2] (1 + Sqrt[2])^n)/(2 Sqrt[2])

Simplified and in standard mathematical notation, this says that there are

1

2

((
1−
√

2
)n+1

+
(

1 +
√

2
)n+1

)

n–words from the alphabet {0, 1, 2} such that neighboring letters differ by at

most 1.

Problem 6. If n people sit around a circular table, how many of the n! ar-

rangements are distinct, i.e., do not have the same neighboring relations?

Solution: The first thing to note about this problem is that it appears to be

ambiguous. Is the neighboring relation who is sitting next to whom, or who is

sitting to the left of whom? Let us solve the problem for the latter case. It

turns out that the final answers simply differ by a factor of 2.

We represent the seating arrangements by the combinatorial class C =

Cyc
(
1
)
. That is, the objects in C are labeled cycles. With Genfunlib, we

can find c[z], the generating function of C, and extract coefficients.

In[73]:= ToGFEqns[Spec[{c == SMCyc[ZClass[1]]}, True], z]

/. z[1] -> z

Out[73]= {c[z] == Log[1/(1 - z)]}

In[74]:= SeriesCoefficient[Log[1/(1 - z)], {z, 0, n}]

Out[74]= 1/n

Since c[z] is an exponential generating function, the number of distinct

arrangements is n!/n = (n− 1)!.

Problem 7. A permutation p of the set {1, . . . , n} is called an involution if

p ◦ p = Id. Find a recursion for the number tn of involutions of {1, . . . , n}. Also

find a closed formula in the form of a sum [9].

Solution: The functional graph of an involution is a set of 1– and 2–cycles.

So we define the class G by

G = Set
(
Cyc≤2

(
1
))
.

With Genfunlib, we find the generating function g[z] for G.

In[75]:= ToGFEqns[Spec[{g == SMSet[SMCyc[ZClass[1]],

Cardinality -> (# <= 2 &)]}, True], z] /. z[1] -> z

Out[75]= {g[z] == 1 + Log[1/(1 - z)] + 1/2 Log[1/(1 - z)]^2}

From this expression for g[z], Mathematica can produce a recurrence rela-

tion for tn, which we call tn.

In[76]:= tn = FullSimplify[ n! SeriesCoefficient[

1 + Log[1/(1 - z)] + 1/2 Log[1/(1 - z)]^2, {z, 0, n}]];

In[77]:= DifferenceRootReduce[tn]

Out[77]= DifferenceRoot[

Function[{y, n}, {n^2 y[n] + (-1 - 2 n) y[ 1 + n] +

y[2 + n] == 0, y[1] == 1, y[2] == 2}]][n]

That is, t1 = 1, t2 = 2, and, for n ≥ 3,

n2tn − (2n+ 1)tn+1 + tn+2 = 0.

Now, finding a sum for tn should be easy since we have the generating func-

tion g[z]. There is a small problem, though, which is that Mathematica tries to

“simplify” the sum into something else as soon as it is obtained. To avoid this, we

replace log(1/(1−z)) in g[z] with the placeholder f(z), then extract coefficients

from g[z] in terms of the coefficients of f(z), using Genfunlib’s enhancements to

Mathematica’s SeriesCoefficient. Then we replace coefficients of f(z) with

the coefficients of log(1/(1− z)), and we are done.

In[78]:= $FullAnalytic = True;

In[79]:= n! SeriesCoefficient[

1 + Log[1/(1 - z)] + 1/2 Log[1/(1 - z)]^2 /.

Log[1/(1 - z)] -> f[z], {z, 0, n}]

Out[79]= n!*(SeriesCoefficient[f[z], {z, 0, n}] +

Sum[SeriesCoefficient[f[z], {z, 0, n - $14}, Assumptions ->

n >= 1 && Element[n, Integers] && Element[$14, Integers] &&

$14 >= 0]*SeriesCoefficient[f[z], {z, 0, $14},

Assumptions -> n >= 1 && Element[n, Integers] &&

Element[$14, Integers] && $14 >= 0], {$14, 0, n}]/2)

In[80]:= % /. SeriesCoefficient[f[z], {z, 0, j_}, ___] :>

Boole[j > 0]/j

Out[80]= n!*(Boole[n > 0]/n + Sum[(Boole[n - $14 > 0]*

Boole[$14 > 0])/ ((n - $14)*$14), {$14, 0, n}]/2)

The result is a bit cluttered, but after a little inspection, it says that, for

n ≥ 1,

tn = n!

 1

n
+

1

2

n−1∑
j=1

1

j(n− j)

 .

Problem 8. Let f(n) be the number of n–words without neighboring zeros

from the alphabet {0, 1, 2}. Find a recursion and a formula for f(n).

Solution: To specify the language of interest, we define the grammar G with

the following productions.

A→ 1 ·A

A→ 2 ·A

A→ 0 ·B

B → 1 ·A

B → 2 ·A

A→ ε

B → ε

In Genfunlib, we can define the corresponding grammar oneZero and com-

pute the generating function.

In[81]:= oneZero = RRGrammar[{

a -> RRGrammarConcat["1", a],

a -> RRGrammarConcat["2", a],

a -> RRGrammarConcat["0", b],

b -> RRGrammarConcat["1", a],

b -> RRGrammarConcat["2", a],

a -> EmptyWord,

b -> EmptyWord}];

In[82]:= GeneratingFunction[ ToRegex[oneZero],

{"0" -> z, "1" -> z, "2" -> z}]

Out[82]= (1 + z)/(1 - 2 z - 2 z^2)

Figure 9 shows a digraph with labeled vertices corresponding to oneZero,

which was created with Genfunlib’s ToDigraph command and edited slightly for

clarity.

Finding a formula for f(n) is now as simple as extracting coefficients using

Mathematica’s SeriesCoefficient command.

In[83]:= SeriesCoefficient[%, {z, 0, n}]

Out[83]= (-2 (1 - Sqrt[3])^n + Sqrt[3] (1 - Sqrt[3])^n +

2 (1 + Sqrt[3])^n + Sqrt[3] (1 + Sqrt[3])^n)/(2 Sqrt[3])

After some simplification, we have that

f(n) =
1

6

(
3 + 2

√
3
)(

1 +
√

3
)n
− 1

6

(
2
√

3 − 3
)(

1−
√

3
)n

.

Obtaining a recurrence relation is as easy as calling a Mathematica command

called DifferenceRootReduce .

In[84]:= DifferenceRootReduce[%, n] // FullSimplify

Out[84]= DifferenceRoot[ Function[{y, n}, {-2 y[n] -

2 y[1 + n] + y[2 + n] == 0, y[0] == 1, y[1] == 3}]][n]

That is, f(0) = 1, f(1) = 3, and, for n ≥ 2,

f(n+ 2) = 2f(n) + 2f(n+ 1).

Figure 9: A digraph with labeled vertices that is equivalent to the grammar
G from Problem 8, which accepts words in {0, 1, 2}∗ that avoid the subword
00. This non–minimal digraph was generated with the ToDigraph command
and plotted using Mathematica’s built–in functionality. Green vertices are start
vertices, and vertices with dotted borders are accepting vertices.

Figure 10: The 12 binary trees with 3 labeled leaves.

Problem 9. How many binary trees with n labeled leaves are there? [9]

Solution:

We assume that “binary trees” are rooted plane trees such that every node

has either 0 or 2 children. We assume that these leaf–labeled binary trees are

binary trees whose leaves are uniquely labeled. The task is to count how many

leaf–labeled binary trees there are with n leaves. Figure 10 shows all leaf–labeled

binary trees with 3 leaves.

First we define a combinatorial specification for the classes of leaves, L, inner

nodes I, and trees T .

L = 1

I = E

T = L+ T × I × T

In Genfunlib, we can convert this specification to equations in the exponen-

tial generating functions l[z] for L, i[z] for I, and t[z] for T .

In[85]:= ToGFEqns[Spec[{

l == ZClass[1],

i == EClass,

t == SMPlus[l, SMTimes[t, i, t]]

}, True], z] /. z[1] -> z

Out[85]= {l[z] == z, i[z] == 1, t[z] == l[z] + i[z] t[z]^2}

We use Mathematica’s Eliminate to provide us with an equation in just our

generating function of interest, t[z].

In[86]:= eqn = Eliminate[%, {l[z], i[z]}]

Out[86]= (1 - t[z]) t[z] == z

The power series equation (1− T (z))T (z) = z has two solutions in general,

but only one with nonnegative coefficients: (1/2)
(
1−
√

1− 4z
)
. Extracting

coefficients with Mathematica’s SeriesCoefficient and multiplying by n! gives

the final answer for the number of leaf–labeled binary trees with n leaves.

In[87]:= $Assumptions = n >= 1;

In[88]:= n! SeriesCoefficient[1/2 (1 - Sqrt[1 - 4 z]),

{z, 0, n}]

Out[88]= -(-1)^n 2^(-1 + 2 n) Binomial[1/2, n] n!

Mathematica reports here that the formula is

−(−1)n22n−1n!

( 1
2

n

)
,

which is just a more cumbersome form of the expected result

(2(n− 1))!

(n− 1)!
.

We note that one way to come up with the specification in this solution is to

find the counting sequence for leaf–labeled binary trees in the Encyclopedia of

Combinatorial Objects at http://algo.inria.fr/ecs. It is entry number 51,

“Labelled Plane Binary Trees”. The Encyclopedia shows the unlabeled Maple–

syntax specification {S = Union(Z,Prod(S,S))}, which we can use to get a

specification in Genfunlib via the ToGenfunlibSpec command.

In[89]:= ToGenfunlibSpec["{S = Union(Z,Prod(S,S))}", False]

Out[89]= Spec[{ss == SMPlus[ZClass[1], SMTimes[ss, ss]]}, False]

Problem 10. Let k and n be integers such that 1 ≤ k ≤ n. Consider all

finite sequences of positive integers with sum n. Suppose that the term k occurs

T (n, k) times in all these sequences. Find T (n, k). [9]

Solution: We start with a specification of the class S of integer compositions

where Z1 marks the number of parts equal to k, and Z2 marks the sum of all

parts.

P1 = MSet≥1,<k(Z2)

P2 = MSetk(Z2)×Z1

P3 = MSet>k(Z2)

S = Seq(P1 + P2 + P3).

Here, P1 is the set of parts smaller than k but no smaller than 1; P2 is the

singleton set containing the part of size k, marked with Z1; and P3 is the

set of parts larger than k. Now, we ultimately wish to count the number of

occurrences of k in all compositions of n. We do this using the symbolic method

by defining a class F each of whose elements corresponds to an occurrence of

k in a composition of n. It suffices to define F as the class obtained by taking

S and pointing separately at each atom Z1, then substituting E for Z1 so that

the objects in F are only weighted by Z2, which marks the total size of the

composition.

F = (ΘS) ◦ E

We give the whole specification to Genfunlib and compute the corresponding

system of ordinary generating function equations. (We assume k ≥ 2 for sim-

plicity; the case k = 1 can easily be handled separately.)

In[90]:= $Assumptions = Element[k, Integers] && k >= 2;

In[91]:= spec = Spec[{

p1 == SMMultiset[ZClass[2], Cardinality -> (1 <= # < k &)],

p2 == SMTimes[ZClass[1],

SMMultiset[ZClass[2], Cardinality -> (# == k &)]],

p3 == SMMultiset[ZClass[2], Cardinality -> (# > k &)],

s == SMSeq[SMPlus[p1, p2, p3]],

f == SMSub[SMPointing[s], EClass]

}, False];

In[92]:= eqns = ToGFEqns[spec, z];

The full system of equations stored in eqns is long, so we do not show it in

full. Let S(z1, z2) be the ordinary generating function for S and let F (z2) be

the ordinary generating function for F .We solve the system to obtain S(z1, z2),

represented in Genfunlib by s[z[1],z[2]].

In[93]:= s[z[1], z[2]] = s[z[1], z[2]] /. Solve[Eliminate[

eqns, {p1[z[1], z[2]], p2[z[1], z[2]], p3[z[1], z[2]]}],

s[z[1], z[2]]] // First

Out[93]= -((-1 + z[2])/( 1 - 2 z[2] + z[2]^k - z[1] z[2]^k -

z[2]^(1 + k) + z[1] z[2]^(1 + k)))

We manually compute F (z2), represented by f[z[2]], using s[z[1]],z[2]]

and the equation

F (z2) = (∂/∂z1)S(z1, z2)|z1=1.

In[94]:= f[z[2]] = D[%, z[1]] /. z[1] -> 1

Out[94]= ((-1 + z[2]) (-z[2]^k + z[2]^(1 + k)))/(1 - 2 z[2])^2

Extracting coefficients from F (z2) with a command built into Mathematica,

we get the final result:

T (n, k) =


2−k+n−2(−k + n+ 3) if k ≤ n− 2,

2n−k if k = n− 1,

2n−k(−k + n+ 1) if k = n.

Problem 11. An n–term sequence (x1, x2, . . . , xn) in which each term is either

0 or 1 is called a binary sequence of length n. Let an be the number of binary

sequences of length n containing no three consecutive terms equal to 0, 1, 0 in

that order. Let bn be the number of binary sequences of length n that contain

no four consecutive terms equal to 0, 0, 1, 1 or 1, 1, 0, 0 in that order. Prove that

bn+1 = 2an for all positive integers n. (USA Mathematical Olympiad 1996 #4)

Solution: We use the letters a and b to represent 0 and 1, respectively, so

n–term binary sequences become n–words over the alphabet {a, b}.

In an actual math contest, it would be most efficient to solve this problem

using a bijection. However, a brute–force solution is feasible if it puts Genfunlib

to work.

We define a regular expression r1 that matches words that avoid aba. In

Genfunlib we represent this regular expression by r1, which is set to the com-

plement of the regular expression (a|b)∗aba(a|b)∗ with respect to the alphabet

{a, b}.

In[95]:= r1 = RegComplement[

ToRegex[RegularExpression["(a|b)*(aba)(a|b)*"]], {"a", "b"}]

Out[95]= Regex[RegexConcat[

RegexStar[RegexOr["b", RegexConcat["a", RegexStar["a"],

"b", "b"]]], RegexOr[EmptyWord, RegexConcat["a",

RegexStar["a"], RegexOr[EmptyWord, "b"]]]]]

We represent the generating function
∑
n≥0 anz

n by f1.

In[96]:= f1 = FullSimplify[GeneratingFunction[r1,

{"a" -> z, "b" -> z}]]

Out[96]:= (1 + z^2)/(1 + z (-2 + z - z^2))

We define r2 as the complement of (a|b)∗aabb(a|b)∗, and we define r3 as the

complement of (a|b)∗bbaa(a|b)∗.

In[97]:= r2 = RegComplement[

ToRegex[RegularExpression["(a|b)*(aabb)(a|b)*"]],

{"a", "b"}];

In[98]:= r3 = RegComplement[

ToRegex[RegularExpression["(a|b)*(bbaa)(a|b)*"]],

{"a", "b"}];

We compute the generating function
∑
n≥0 bnz

n, represented by f2, by tak-

ing the intersection of r2 and r3.

In[99]:= f2 = GeneratingFunction[

RegIntersection[r2, r3], {"a" -> z, "b" -> z}]

// FullSimplify

Out[99]= (1 + z^2 + z^3)/(1 + z (-2 + z - z^2))

Now, we can conclude by showing that f2 - 2*z*f1 is a constant.

In[100]:= f2 - 2 z*f1 // FullSimplify

Out[100]= 1

5.2 Miscellaneous problems

Problem 12. A word w avoids a subword pattern p if and only if no contiguous

subword (a.k.a factor) of w is equal to p. Let fk(n) be the number of n–words

over the alphabet {0, 1} that avoid the subword pattern 1k = 11 · · · 1. Find

Fk(z) =
∑
n≥0 fk(n)zn.

Solution: This is a generalization of Problem 2, which is the case k = 3. We

cannot define automata with parameters such as k, but we can use a recurrence

relation to solve this problem instead. An n–word w ∈ {0, 1}∗ avoids 1k if and

only if w = 1j0w′, where 0 ≤ j < k and w′ is a binary (n − j − 1)–word that

avoids 1k. If we define f(n) = 0 for all n < 0, this leads to a recurrence relation

for fk(n) :

fk(n) =
k∑
i=1

fk(n− i) + [n < k], for all n ∈ Z.

The [n < k] term accounts for the words 1n, where n < k.

Now we write this equation out in Mathematica using f[n] to represent f(n),

and apply the Genfunlib–enhanced GeneratingFunction command to perform

the generating function transform to both sides.

In[101]:= $Assumptions =

Element[i, Integers] && i >= 1 && Element[k, Integers]

&& k >= 1;

In[102]:= eqn = f[n] == Sum[f[n - i], {i, 1, k}] + Boole[n < k];

In[103]:= GeneratingFunction[eqn, n, z]

Out[103]= GeneratingFunction[f[n], n, z] == (-1 + z^k)/(-1 + z) +

Sum[GeneratingFunction[f[-i + n], n, z], {i, 1, k}]

For some reason, this output must be evaluated again to get the desired result.

In[104]:= GeneratingFunction[f[n], n, z] == (-1 + z^k)/(-1 + z) +

Sum[GeneratingFunction[f[-i + n], n, z], {i, 1, k}]

Out[104]= GeneratingFunction[f[n], n, z] == (-1 + z^k)/(-1 + z) +

Sum[z^i*GeneratingFunction[f[n], n, z] + Sum[z^(i + n)*f[n],

{n, -i, -1}], {i, 1, k}]

We can replace GeneratingFunction[f[n], n, z], which represents F (z),

with the shorter expression ff[z]; and we can remove remaining f[n] terms

with zero, because they are all equal to zero.

In[105]:= % /. {GeneratingFunction[f[n], n, z] -> ff[z], f[n] -> 0}

Out[105]= ff[z] == (-1 + z^k)/(-1 + z) + (z (-1 + z^k) ff[z])/(-1 + z)

In mathematical notation, this is

F (z) =
zF (z)

(
zk − 1

)
z − 1

+
zk − 1

z − 1
.

All that remains is to solve this linear equation.

In[106]:= ff[z] /. Solve[%, ff[z]] // First

Out[106]= (1 - z^k)/(1 - 2 z + z^(1 + k))

So the final answer is

F (z) =
1− zk

zk+1 − 2z + 1
.

Problem 13. Let f(n) be the number of n–words without neighboring zeros

from the alphabet {0, 1, 2}. Find an asymptotic expression for f(n) as n goes

to infinity.

Solution: This f(n) is the same as the one from Problem 8, where we discov-

ered that the ordinary generating function for f(n) is

∑
n≥0

f(n)zn =
1 + z

1− 2z − 2z2
.

The required asymptotic expression is extracted by simply applying Genfunlib’s

CoefLimit command to the generating function.

In[107]:= CoefLimit[(1 + z)/(1 - 2 z - 2 z^2), z, n]

Out[107]= ((1/2 (-1 + Sqrt[3]))^(-1 -

n) (1 + 1/2 (-1 + Sqrt[3])))/(2 + 2 (-1 + Sqrt[3]))

After automatic simplification, this says

f(n) ∼ 1

3

(
3 +
√

3
)

2n−1
(√

3 − 1
)−n−1

, as n→∞.

Problem 14. Define the class of plane trees, where Z1 marks nodes and Z2

marks path length, by

T = Z1 × Seq (T ◦ (Z1 ×Z2)) .

The path length of a tree is the sum of the distances of all nodes to the root.

1. Find an asymptotic expression for the number of plane trees with n nodes,

as n goes to infinity.

2. Find the number of plane trees with 7 nodes and pathlength 8.

3. Find an asymptotic expression for the average path length of a plane tree

with n nodes, as n goes to infinity.

Solution: To solve part 1, we remove Z2 from the specification and use Gen-

funlib’s ToGFEqns command to get an equation in T (z1), which is represented

by t[z[1]].

In[108]:= spec = Spec[{

t == SMTimes[ZClass[1], SMSeq[t]]

}, False];

In[109]:= treeEqn = First[ToGFEqns[spec, z]]

Out[109]= t[z[1]] == z[1]/(1 - t[z[1]])

The returned equation (treeEqn), which is t(z1) = z1/(1− t(z1)), can be given

to Genfunlib’s TreeAsymptot command to compute an asymptotic expression

for [zn1 ]t(z1).

In[110]:= TreeAsymptot[treeEqn, n]

Out[110]= 4^n ((1/n)^(3/2)/(4 Sqrt[Pi])+O[1/n]^(5/2))

Converted to LATEX, this says that the number of plane trees with n nodes is

[zn1 ]T (z1) = 4n

((
1
n

)3/2
4
√
π

+O(n−5/2)

)
.

Now, we assign the full specification, including Z2, to the augSpec variable,

and save the corresponding bivariate ordinary generating function equation in

the eqn variable.

In[111]:= augSpec = Spec[{

t == SMTimes[ZClass[1],

SMSeq[SMSub[t, SMTimes[ZClass[1], ZClass[2]]]]

] }, False];

In[112]:= eqn = First[ToGFEqns[augSpec, z]]

Out[112]= t[z[1], z[2]] == z[1]/(1 - t[z[1] z[2], z[2]])

That is, we have

T (z1, z2) =
z1

1− T (z1z2, z2)
. (2)

Equation (2) is all we need to answer part 2, since we can pass it to Genfunlib’s

CoefsByDerivs command to compute the initial terms of t[z[1], z[2]].

In[113]:= CoefsByDerivs[eqn, t[z[1], z[2]], {z[1], 0, 7},

{z[2], 0, 8}]

Out[113]= {t[z[1],z[2]]==O[z[2]]^9+(1+O[z[2]]^9) z[1]+

(z[2]+O[z[2]]^9) z[1]^2+ (z[2]^2+z[2]^3+O[z[2]]^9) z[1]^3+

(z[2]^3+2 z[2]^4+z[2]^5+z[2]^6+O[z[2]]^9) z[1]^4+

(z[2]^4+3 z[2]^5+3 z[2]^6+3 z[2]^7+2 z[2]^8+

O[z[2]]^9) z[1]^5+(z[2]^5+4 z[2]^6+6 z[2]^7+

7 z[2]^8+O[z[2]]^9) z[1]^6+(z[2]^6+5 z[2]^7+10 z[2]^8+

O[z[2]]^9) z[1]^7+O[z[1]]^8}

This says [z71z
8
2 ]T (z1, z2) = 10.

For part 3, we need to compute (an asymptotic estimate of)

[zn1 ]T (z1, 1)

[zn1 ] ∂
∂z2

T (z1, z2)
∣∣∣
z2=1

.

We set z2 to 1 on both sides of Equation (2), and save the result in eqn1.

Then we solve that equation for T (z1, 1) and store the solution as t00.

In[114]:= eqn1 = eqn /. z[2] -> 1;

In[115]:= t00 = First[t[z[1], 1] /. Solve[eqn1, t[z[1], 1]]]

Out[115]= 1/2 (1 - Sqrt[1 - 4 z[1]])

We define eqn2 as the equation obtained by differentiating both sides of

Equation (2) with respect to z2, and setting z2 to 1.

In[116]:= eqn2 = D[eqn, z[2]] /. z[2] -> 1;

We define eqn3 as the equation obtained by differentiating both sides of Equa-

tion 2 with respect to z1.

In[117]:= eqn3 = D[eqn1, z[1]];

Finally, we define eqn4 as the equation obtained by eliminating T (z1, 1) and

∂
∂z1

T (z1, 1) from the system eqn1 && eqn2 && eqn3.

In[118]:= eqn4 = Eliminate[

eqn1 && eqn2 && eqn3, {t[z[1], 1],

Derivative[1, 0][t][z[1], 1]}];

As eqn1 is for T (z1, 1), eqn4 is a single equation in which the only unknown

is ∂
∂z2

T (z1, z2)
∣∣∣
z2=1

. We solve eqn4 for ∂
∂z2

T (z1, z2)
∣∣∣
z2=1

and save the solution

in t01.

In[119]:= t01 = PowerExpand[ Simplify[First[(t^(0, 1))[z[1], 1] /.

Solve[eqnt4, Derivative[0, 1][t][z[1], 1]]]]];

Out[119]= (z[1] - 4 z[1]^2 - I z[1] (-1 + 4 z[1])^(3/2))/

(2 (1 - 4 z[1])^2)

Now, Genfunlib does not have the ability to obtain an asymptotic expression

for the coefficients of t00 or t01, but Mathematica does. We define the function

coefAsympt which uses Mathematica’s Series and SeriesCoefficient com-

mands to extract an expression for coefficient n of a power series, and convert

to an asymptotic form.

In[120]:= $Assumptions = n > 1 && Element[n, Integers];

In[121]:= coefAsympt[series_, var_] := FullSimplify[

Series[SeriesCoefficient[series, {var, 0, n}],

{n, Infinity, 1}]];

All that remains is to apply coefAsympt to t00 and t01 and divide, obtaining

the answer to part 3.

In[122]:= coefAsympt[t01,z[1]]/coefAsympt[t00,z[1]]

Out[122]= 1/2 Sqrt[Pi] n^(3/2)-n/2+1/Sqrt[O[1/n]]

That is, as n goes to infinity,

[zn1 ]T (z1, 1)

[zn1 ] ∂
∂z2

T (z1, z2)
∣∣∣
z2=1

=
1

2

√
π n3/2 − n

2
+O

(√
n
)
.

6 Further work

In terms of further work, there is a number of possible directions. As mentioned

previously, Mathematica itself causes difficulties when one wants to work with

general big-O expressions, iterated sums with a symbolic number of dummy

variables, and recurrence relations with polynomial coefficients. To get around

these issues, perhaps a package could be written to provide alternatives or en-

hancements to the core Mathematica commands involved. We also mentioned

that performace in Genfunlib should be improved across the board, and specif-

ically in the area of regular languages. Multi-dimensional Newton iteration

for computing coefficients from transcendental power series equations could be

added.

Many counting problems involve not regular languages but context-free lan-

guages. Context-free languages do not have all the decidable properties that

regular languages do, e.g. testing whether a context-free grammar is not decid-

able, and neither is testing whether a context-free grammar has an equivalent

deterministic context-free grammar. However, if grammars are simply assumed

to be unambiguous, something could be built, similar to Genfunlib’s symbolic

method functionality, to obtain generating functions that count the number of

words of length n in a context-free language.

Genfunlib focuses on counting combinatorial objects, but there are also the

problems of enumeration and random generation. For families of objects repre-

sented by combinatorial species, a recent work by Pivoteau et. al [26] presents

algorithms based on combinatorial Newton iteration [1] that are relevant to both

of these problems. So far, the algorithm for random generation with species us-

ing Boltzmann sampling has not been implemented.

Extracting an asymptotic estimate for the coefficients [zn]F (z) of a power

series is an interesting problem that Genfunlib approaches only on a relatively

basic level. Algorithms that handle a larger set of power series F (z) are needed.

In addition, algorithms and implementations for multivariate asymptotics would

be useful. Some results in this area have been found [24, 28], and Raichev’s amgf

package for Sage implements a method for computing the asymptotic expansion

of

[zrα1
1 zrα2

2 · · · ]F (z1, z2, . . . ),

as r →∞, for a certain class of multivariate power series F (z1, z2, . . . ).

Appendix

Asymptotics.m

(* Mathematica Package *)

BeginPackage["Genfunlib`Asymptotics`"]

TreeAsymptot::usage = "TreeAsymptot[y[z]==z*phi,n] gives an asymptotic " <>

"series for SeriesCoefficient[y[z], {z,0,n}] as " <>

"n goes to infinity, where y[z] is the solution to y[z]==z*phi that is " <>

"analytic at 0.";

CoefLimit::usage = "CoefLimit[f,z,n] gives an expression that is " <>

"asymptotically equivalent to SeriesCoefficient[f, {z,0,n}] as n goes to " <>

"infinity.";

Begin["`Private`"]

(* define GCD and Mod when Infinity is an argument *)

protected = Unprotect[GCD, Mod];

GCD[Infinity, rest__] := GCD[rest];

GCD[0] := Infinity;

Mod[n_, Infinity] := n;

Protect[Evaluate[protected]];

(* validation of equations given as input to asymptotic functions *)

TreeAsymptot::invalid = "Not an elementary functional equation.";

CoefLimit::invalid = "Not an elementary function.";

TreeAsymptot::invalidArgumentSyntax = "Invalid argument syntax.";

CoefLimit::invalidArgumentSyntax = "Invalid argument syntax.";

(* what can usually be used as a variable, according to

ref/message/General/ivar *)

variablePattern = Except[_String | _?NumberQ | _Plus | _Times |

_Sum | _Product | _^_Integer];

validatePhi[expr_, y_[z_]] := Module[

{

ok = True

},

(* must not be deg 1 polynomial *)

ok = ok && !(PolynomialQ[expr, y[z]] && Exponent[expr, y[z]] <= 1);

ok = ok && SeriesCoefficient[expr, {y[z], 0, 0}] != 0;

(* and elementary function *)

ok = ok && ea[expr, y[z]];

If[ ok =!= True, Message[TreeAsymptot::invalid]];

ok

];

validateExpr[expr_, x_] := Module[

{

ok = True

},

ok = ok && ea[expr, x];

If[ ok =!= True, Message[CoefLimit::invalid]];

ok

];

(* ea tests whether an expression is an "elementary function" *)

ea[expr_, x_] /; PolynomialQ[expr, x] := Module[

{

ok = True

},

ok = ok &&

And @@ ((Element[#, Rationals] && # >= 0) & /@ CoefficientList[expr, x]);

ok = ok && CoefficientList[expr, x] != {0};

ok

];

ea[1/(1 + arg_), x_] := Module[

{

ok = True

},

ok = ok && ea[-arg, x];

ok = ok && (Limit[arg, x -> 0] === 0);

ok

];

ea[Log[1/(1 + arg_)], x_] := Module[

{

ok = True

},

ok = ok && ea[-arg, x];

ok = ok && (Limit[arg, x -> 0] === 0);

ok

];

ea[Exp[arg_], x_] := Module[

{

ok = True

},

ok = ok && ea[arg, x];

ok = ok && (Limit[arg, x -> 0] === 0);

ok

];

ea[arg1_ + arg2_, x_] := ea[arg1, x] && ea[arg2, x];

ea[arg1_ * arg2_, x_] := ea[arg1, x] && ea[arg2, x];

ea[Power[expr_, j_Integer?Positive], x_] := ea[expr, x];

ea[Power[expr_, j_Integer?(#<-1&)], x_] := ea[1/expr, x];

ea[___] := False;

exponents[y_, x_] :=

Union[Flatten[CoefficientRules[y, x][[All, 1]]]];

(* reduction implements the algorithm in Flajolet et al. (Automatic *)

(* average-case analysis of algorithms) on p78 *)

reduction[y_, x_] := Module[

{exps = exponents[y, x]},

(

If[Length[exps] == 1,

{First[exps], Infinity},

{Mod[First[exps], GCD @@ Differences[exps]],

GCD @@ Differences[exps]}

]

) /; PolynomialQ[y, x]

];

reduction[y : 1/(1 + arg_), x_] := Module[

{

arga, argp

},

{arga, argp} = reduction[-arg, x];

{0, GCD[arga, argp]}

83



];

reduction[y : Log[1/(1 + arg_)], x_] := Module[

{

arga, argp

},

{arga, argp} = reduction[-arg, x];

{0, GCD[arga, argp]}

];

reduction[y : Exp[arg_], x_] := Module[

{

arga, argp

},

{arga, argp} = reduction[arg, x];

{0, GCD[arga, argp]}

];

(* some details of this algorithm in Flajolet et al. were left out. to prove *)

(* that the following is correct, *)

(* give equivalent definitions of a and p in terms of GCD and Mod, *)

(* show that the following supplies that; note: term1a=term2a (mod pp) *)

reduction[term1_ + term2_, x_] := Module[

{

term1a, term1p, term2a, term2p,

pp

},

{term1a, term1p} = reduction[term1, x];

{term2a, term2p} = reduction[term2, x];

pp = GCD[term1p, term2p, term1a - term2a];

{Mod[term1a, pp], pp}

];

reduction[term1_ * term2_, x_] := Module[

{

term1a, term1p, term2a, term2p,

pp

},

{term1a, term1p} = reduction[term1, x];

{term2a, term2p} = reduction[term2, x];

pp = GCD[term1p, term2p];

{Mod[term1a + term2a, pp], pp}

];

reduction[Power[expr_, j_Integer?Positive], x_] := reduction[Unevaluated[

expr * Power[expr, j - 1]], x];

reduction[Power[expr_, j_Integer?(#<-1&)], x_] := reduction[Unevaluated[

Power[expr, -1] * Power[expr, j + 1]], x];

(* assumes expr is from elementary iterative class, so characteristic \

eqn is satisfied and period is computed *)

TreeAsymptot[(y:variablePattern)[z:variablePattern] ==

(z:variablePattern) * expr_, n:variablePattern] := Module[

{

period = reduction[expr, y[z]], p,

t, res, nres,

d1, r

},

(

p = If[period[[2]] == Infinity,

period[[1]], period[[2]]];

(* solve characteristic equation *)

res = Solve[expr - y[z] * D[expr, y[z]] == 0 && y[z] > 0, y[z]];

If[MatchQ[res, {__List}],

t = Min[y[z] /. res],

nres =

NSolve[expr - y[z]*D[expr, y[z]] == 0 && y[z] > 0, y[z]];

t = Min[y[z] /. nres]

];

r = t / (expr /. y[z] -> t);

d1 = Sqrt[(2 * (expr /. y[z] -> t))/(

D[expr, {y[z], 2}] /. y[z] -> t)];

If[p >= 2,

Boole[Mod[n, p] == 1] * r ^ -n *

SeriesData[n, DirectedInfinity[1], {(p*d1)/(2*Sqrt[Pi]), 0},

3, 5, 2],

r^-n*

SeriesData[n, DirectedInfinity[1], {d1/(2*Sqrt[Pi]), 0}, 3,

5, 2]

]

) /; validatePhi[expr, y[z]]

];

TreeAsymptot[___] /; (Message[TreeAsymptot::invalidArgumentSyntax]; False) :=

Null;

(* radius implements the algorithm from Flajolet et al. on p77 *)

radius[y_, x_] /; PolynomialQ[y, x] := Infinity;

radius[Exp[y_], x_] := radius[y, x];

radius[1/(1 + y_), x_] := Module[

{

res, r, nres

},

res = Solve[-y == 1 && x > 0, x];

r = If[MatchQ[res, {__List}],

Min[x /. res],

nres = NSolve[-y == 1 && x > 0, x];

Min[x /. nres]

];

r

];

radius[Log[1/(1 + y_)], x_] := Module[

{

res, r, nres

},

res = Solve[-y == 1 && x > 0, x];

r = If[MatchQ[res, {__List}],

Min[x /. res],

nres = NSolve[-y == 1 && x > 0, x];

Min[x /. nres]

];

r

];

radius[term1_ + term2_, x_] := Min[radius[term1, x], radius[term2, x]];

radius[factor1_*factor2_, x_] := Min[radius[factor1, x], radius[factor2, x]];

radius[Power[expr_, j_Integer?Positive], x_] := radius[Unevaluated[

expr * Power[expr, j - 1]], x];

radius[Power[expr_, j_Integer?(#<-1&)], x_] := radius[Unevaluated[

Power[expr, -1] * Power[expr, j + 1]], x];

(* directions implements the algorithm from Flajolet et al. on p79 *)

directions[Exp[expr_], x_] := directions[expr, x];

directions[1/(1 + expr_), x_] := Module[

{

a, p, q

},

{a, p} = reduction[-expr, x];

q = GCD[a, p];

2*Range[1, q]*Pi/q

];

directions[Log[1/(1 + expr_)], x_] := Module[

{

a, p, q

},

{a, p} = reduction[-expr, x];

q = GCD[a, p];

2*Range[1, q]*Pi/q

];

directions[Plus[term1_, term2_], x_] := Module[

{

r1 = radius[term1, x],

r2 = radius[term2, x]

},

If[r1 < r2,

Return[directions[term1, x]]

];

If[r1 > r2,

Return[directions[term2, x]]

];

Return[Union[directions[term1, x], directions[term2, x]]];

];

directions[Times[term1_, term2_], x_] := Module[

{

r1 = radius[term1, x],

r2 = radius[term2, x]

},

If[r1 < r2,

Return[directions[term1, x]]

];

If[r1 > r2,

Return[directions[term2, x]]

];

Return[Union[directions[term1, x], directions[term2, x]]];

];

directions[Power[expr_, j_Integer?Positive], x_] := directions[Unevaluated[

expr * Power[expr, j - 1]], x];

directions[Power[expr_, j_Integer?(#<-1&)], x_] := directions[Unevaluated[

Power[expr, -1] * Power[expr, j + 1]], x];

(* allimit implements the Expansion algorithm from Flajolet et al. on p81. *)

(* it returns a 4-tuple {a, b, c, d}, where the expansion is *)

(* a*(1-z/d)^b*Log[1-z/d]^c *)

allimit[1/(1 + expr_), x_] := Module[

{

r = radius[1/(1 + expr), x]

},

{1/(r*D[-expr, x] /. x -> r), -1, 0, r}

];

allimit[Log[1/(1 + expr_)], x_] := Module[

{

r = radius[Log[1/(1 + expr)], x]

},

{1, 0, 1, r}

];

allimit[Exp[expr_], x_] := Module[

{

expra, exprb, exprc, exprr,

r = radius[Exp[expr], x]

},

{exprc, expra, exprb, exprr} = allimit[expr, x];

If[MatchQ[{exprc, expra, exprb, exprr}, {_, 0, 1, r}],

{Exp[exprb], -expra, 0, r},

Throw["experr"]

]

];

allimit[term1_ + term2_, x_] := Module[

{

term1limit, term1a, term1b, term1c, term1r,

term2limit, term2a, term2b, term2c, term2r

},

term1limit = Catch[allimit[term1, x]];

term2limit = Catch[allimit[term2, x]];

If[term1limit == term2limit == "experr",

Throw["experr"]

];

If[term1limit == "experr",

If[term2limit[[4]] < radius[term1, x],

Return[term2limit],

Throw["experr"]

]

];

If[term2limit == "experr",

If[term1limit[[4]] < radius[term2, x],

Return[term1limit],

Throw["experr"]

]

];

{term1c, term1a, term1b, term1r} = term1limit;

{term2c, term2a, term2b, term2r} = term2limit;

If[term1r < term2r,

Return[term1limit];

];

If[term2r > term1r,

Return[term1limit];

];

If[term1a < term2a,

Return[term1limit];

];

If[term2a < term1a,

Return[term2limit];

];

If[term1b > term2b,

Return[term1limit];

];

If[term2b > term1b,

Return[term2limit];

];

{term1c + term2c, term1a, term1b, term1r}

];

allimit[factor1_ * factor2_, x_] := Module[

{

factor1limit, factor1a, factor1b, factor1c, factor1r,

factor2limit, factor2a, factor2b, factor2c, factor2r

},

factor1limit = Catch[allimit[factor1, x]];

factor2limit = Catch[allimit[factor2, x]];

If[factor1limit == factor2limit == "experr",

92



Throw["experr"]

];

If[factor1limit == "experr",

If[factor2limit[[4]] < radius[factor1, x],

{factor2c, factor2a, factor2b, factor2r} = factor2limit;

Return[{factor2c * Limit[factor1, x -> factor2r], factor2a, factor2b,

factor2r}],

Throw["experr"]

]

];

If[factor2limit == "experr",

If[factor1limit[[4]] < radius[factor2, x],

{factor1c, factor1a, factor1b, factor1r} = factor1limit;

Return[{factor1c * Limit[factor2, x -> factor1r], factor1a, factor1b,

factor1r}],

Throw["experr"]

]

];

{factor1c, factor1a, factor1b, factor1r} = factor1limit;

{factor2c, factor2a, factor2b, factor2r} = factor2limit;

If[factor1r < factor2r,

Return[{factor1c*(factor2 /. x -> factor1r), factor1a, factor1b,

factor1r}];

];

If[factor2r < factor1r,

Return[{(factor1 /. x -> factor2r)*factor2c, factor2a, factor2b,

factor2r}];

];

{factor1c*factor2c, factor1a + factor2a, factor1b + factor2b, factor1r}

];

allimit[Power[expr_, j_Integer?Positive], x_] := allimit[Unevaluated[

expr * Power[expr, j - 1]], x];

allimit[Power[expr_, j_Integer?(#<-1&)], x_] := allimit[Unevaluated[

Power[expr, -1] * Power[expr, j + 1]], x];

allimit[poly_, x_] /; PolynomialQ[poly,x] := {poly /. x-> Infinity,

0, 0, Infinity};

CoefLimit[expr_, x:variablePattern, n:variablePattern] := Module[

{

r = radius[expr, x],

dirs = directions[expr, x],

expra, exprb, exprc, exprr, exprlimit

},

(

If[dirs != {2 * Pi} || r == Infinity,

Return[SeriesCoefficient[expr, {x, 0, n}]];

];

exprlimit = Catch[allimit[expr, x]];

If[exprlimit == "experr",

Return[SeriesCoefficient[expr, {x, 0, n}]]

];

{exprc, expra, exprb, exprr} = exprlimit;

If[expra < 0,

exprc*exprr^-n (n^-expra Log[n]^exprb)/(n*Gamma[-expra]),

Return[SeriesCoefficient[expr, {x, 0, n}]];

]

) /; validateExpr[expr, x]

];

CoefLimit[___] /; (Message[CoefLimit::invalidArgumentSyntax]; False) := Null;

End[]

EndPackage[]

GFeqn2rec.m

(* Mathematica Package *)

BeginPackage["Genfunlib`GFeqn2rec`"]

(* this file defines no public functions, however it does *)

(* override SeriesCoefficient *)

$FullAnalytic::usage = "$FullAnalytic is True iff SeriesCoefficient should " <>

"assume that " <>

"all subexpressions of the expression to be expanded are " <>

"analytic, when the " <> "expansion is around 0.";

Begin["`Private`"]

protected = Unprotect[SeriesCoefficient];

$FullAnalytic = False;

(* what can usually be used as a variable, according to

ref/message/General/ivar *)

variablePattern = Except[_String | _?NumberQ | _Plus | _Times |

_Sum | _Product | _^_Integer];

arg2 = {x:variablePattern, 0, n_};

(* coefficients of negative or noninteger powers are 0 *)

SeriesCoefficient[_, arg2, opts:OptionsPattern[]] := Module[

{},

0 /; NumericQ[n] && (n < 0 || !IntegerQ[n]) && $FullAnalytic

];

(* addition *)

SeriesCoefficient[expr1_ + expr2_, arg2, opts:OptionsPattern[]] :=

SeriesCoefficient[expr1, {x, 0, n}, opts] + SeriesCoefficient[expr2,

{x, 0, n}, opts];

(* scaling (can produce simpler expressions than general multiplication) *)

SeriesCoefficient[fac_ * expr_, arg2, opts:OptionsPattern[]] /; FreeQ[fac, x] &&

!FreeQ[expr, x] :=

fac * SeriesCoefficient[expr, {x, 0, n}, opts];

(* shifting (can produce simpler expressions than multiplication) *)

SeriesCoefficient[x_^(k_) * expr_, arg2,

opts:OptionsPattern[]] := Module[

{},

(

Boole[n - k >= 0] * SeriesCoefficient[expr,

{x, 0, n - k}, opts]

) /; !FreeQ[expr, x] && $FullAnalytic && FreeQ[k, x]

];

(* multiplication *)

(* doesn't seem to interfere with things like E^x*1/(1-x) *)

SeriesCoefficient[expr1_ * expr2_, arg2, opts:OptionsPattern[]] := With[

{

iterator = Unique[]

},

Module[{

newOpts = Sequence@@(DeleteCases[{opts}, Verbatim[Rule][Assumptions, _]] ~

Append ~ (Assumptions ->

OptionValue[Assumptions] && Element[iterator, Integers] && iterator >= 0))

},

Sum[SeriesCoefficient[expr1, {x, 0, iterator}, newOpts]

* SeriesCoefficient[expr2, {x, 0, n - iterator}, newOpts],

{iterator, 0, n}]

] /; !FreeQ[expr1, x] && !FreeQ[expr2, x] && $FullAnalytic

];

(* derivatives and integrals *)

SeriesCoefficient[Derivative[k_][a_Symbol][(j_:1) * x_], arg2,

opts:OptionsPattern[]] :=

Pochhammer[n + 1, k] j^n SeriesCoefficient[a[x], {x, 0, n + k}, opts] /;

Simplify[Implies[OptionValue[Assumptions], k > 0]] && FreeQ[j, x];

SeriesCoefficient[Derivative[0, k2_][a_Symbol][j_ * x_, expr_], arg2,

opts:OptionsPattern[]] :=

j^n SeriesCoefficient[Derivative[0, k2][a][x, expr], {x, 0, n}, opts] /;

Simplify[Implies[OptionValue[Assumptions], k2 > 0]] && FreeQ[expr, x] &&

FreeQ[j, x];

SeriesCoefficient[Derivative[k1_, 0][a_Symbol][expr_, j_ * x_], arg2,

opts:OptionsPattern[]] :=

j^n SeriesCoefficient[Derivative[k1, 0][a][expr, x], {x, 0, n}, opts] /;

Simplify[Implies[OptionValue[Assumptions], k1 > 0]] && FreeQ[expr, x] &&

FreeQ[j, x];

SeriesCoefficient[Derivative[k1_, k2_][a_Symbol][(j_:1) * x_, expr_], arg2,

opts:OptionsPattern[]] :=

Pochhammer[n + 1, k1] * j^n * SeriesCoefficient[

Derivative[0, k2][a][x, expr], {x, 0, n + k1}, opts] /;

Simplify[Implies[OptionValue[Assumptions], k1 > 0 && k2 >= 0]] &&

FreeQ[expr, x] && FreeQ[j, x];

SeriesCoefficient[Derivative[k2_, k1_][a_Symbol][expr_, (j_:1) * x_], arg2,

opts:OptionsPattern[]] :=

Pochhammer[n + 1, k1] * j^n * SeriesCoefficient[

Derivative[k2, 0][a][expr, x], {x, 0, n + k1}, opts] /;

Simplify[Implies[OptionValue[Assumptions], k1 > 0 && k2 >= 0]] &&

FreeQ[expr, x] && FreeQ[j, x];

SeriesCoefficient[Derivative[k1_, k2_][a_Symbol][(j_:1) * x_, (k_:1) * x_],

arg2, opts:OptionsPattern[]] :=

With[

{iterator = Unique[], first = Unique[], second = Unique[]},

(

Sum[k^iterator * j^(n - iterator) *

Pochhammer[iterator + 1, k1] * Pochhammer[n - iterator + 1, k2]

* SeriesCoefficient[a[first, second], {first, 0, iterator + k1},

{second, 0, n - iterator + k2}, opts], {iterator, 0, n}]

) /; $FullAnalytic &&

Simplify[Implies[OptionValue[Assumptions], k1 >= 0 && k2 >= 0]] &&

FreeQ[j, x] && FreeQ[k, x];

];

SeriesCoefficient[Integrate[expr_,

{t_, 0, x_}], arg2, opts:OptionsPattern[]] := Module[

{},

(

Piecewise[{

{

(1/n) * SeriesCoefficient[expr /. t -> x, {x, 0, n - 1}, opts],

n >= 1

}

}]

) /; $FullAnalytic

];

(* powers of functions *)

SeriesCoefficient[a_Symbol[args__]^(k_Integer?Positive), arg2,

opts:OptionsPattern[]] /; !FreeQ[{args}, x] :=

SeriesCoefficient[

Unevaluated[a[args] * a[args]^(k - 1)],

{x, 0, n}, opts

];

(* compositions *)

SeriesCoefficient[a_Symbol[k_ * x_], arg2, opts:OptionsPattern[]] /;

FreeQ[k, x] := k^n * SeriesCoefficient[a[x], {x, 0, n}, opts];

SeriesCoefficient[a_Symbol[(k_:1) * x_, (j_:1) * x_], arg2,

opts:OptionsPattern[]] := With[

{iterator = Unique[], first = Unique[], second = Unique[]},

(

Sum[k^iterator * j^(n - iterator) *

SeriesCoefficient[a[first, second], {first, 0, iterator},

{second, 0, n - iterator}, opts], {iterator, 0, n}]

) /; $FullAnalytic && FreeQ[k, x] && FreeQ[j, x]

];

SeriesCoefficient[a_Symbol[k_*x_, j_], arg2, opts:OptionsPattern[]] /;

FreeQ[j, x] && FreeQ[k, x] :=

k^n SeriesCoefficient[a[x, j], {x, 0, n}, opts];

SeriesCoefficient[a_Symbol[j_, k_*x_], arg2, opts:OptionsPattern[]] /;

FreeQ[j, x] && FreeQ[k, x] :=

k^n SeriesCoefficient[a[j, x], {x, 0, n}, opts];

(* misc *)

SeriesCoefficient[Sum[expr_, {i_, lb_, ub_}], arg2, opts:OptionsPattern[]] /;

FreeQ[lb, x] && FreeQ[ub, x] :=

Sum[SeriesCoefficient[expr, {x, 0, n}, opts], {i, lb, ub}];

SeriesCoefficient[expr_, {x:variablePattern, 0, n_}, iters__List,

opts:OptionsPattern[]] :=

SeriesCoefficient[SeriesCoefficient[expr, {x, 0, n}, opts], iters,

opts];

Protect[Evaluate[protected]];

End[]

EndPackage[]

RegularLanguages.m

(* Mathematica Package *)

BeginPackage["Genfunlib`RegularLanguages`"]

NFA::usage = "NFA[n, alphabet, transitions, acceptStates, initialState] is " <>

"an NFA " <> "with states [1..n], and transitions given by a matrix.";

DFA::usage = "DFA[n, alphabet, transitions, acceptStates, initialState] is " <>

"a DFA " <> "with states [1..n], and transitions given by a matrix.";

Regex::usage = "Regex is a wrapper for expressions built with strings, " <>

"RegexOr, " <> "RegexConcat, RegexStar, and EmptyWord.";

RRGrammar::usage = "RRGrammar[{lhs -> rhs, ...}] is a right regular " <>

"grammar given " <> "by a list of productions.";

Digraph::usage = "Digraph[graph, startVertices, endVertices, " <>

"\[Epsilon]Accepted] " <>

"is a directed graph with labeled vertices for counting " <>

"walks " <> "from start vertices to end vertices.";

ToNFA::usage = "ToNFA[DFA[...]] is an NFA defined from a DFA.\n" <>

"ToNFA[Regex[...]] is an NFA defined from a symbolic " <>

"regular expression.\n" <>

"ToNFA[RRGrammar[...]] is an NFA defined from a right " <>

"regular grammar.\n" <>

"ToNFA[Digraph[...]] is an NFA defined from a digraph with " <>

"labeled vertices.";

ToDFA::usage = "ToDFA[NFA[...]] is a DFA defined from a NFA.\n" <>

"ToDFA[Regex[...]] is a DFA defined from a symbolic " <>

"regular expression.\n" <>

"ToDFA[RRGrammar[...]] is a DFA defined from a right " <>

"regular grammar.\n" <>

"ToDFA[Digraph[...]] is a DFA defined from a digraph with " <>

"labeled vertices.";

ToRegex::usage = "ToRegex[NFA[...]] is a symbolic regular expression " <>

"defined from an NFA.\n" <>

"ToRegex[DFA[...]] is a symbolic regular expression defined from a DFA.\n" <>

"ToRegex[RegularExpression[\"regex\"]] is a symbolic regular expression " <>

"defined from a restricted Mathematica regular expression.\n" <>

"ToRegex[RRGrammar[...]] is a symbolic regular expression defined " <>

"from a right regular grammar.\n" <>

"ToRegex[Digraph[...]] is a symbolic regular expression defined from a " <>

"digraph with labeled vertices.";

ToRRGrammar::usage = "ToRRGrammar[NFA[...]] is a right regular grammar " <>

"defined from a NFA.\n" <>

"ToRRGrammar[DFA[...]] is a right regular grammar defined from a DFA.\n" <>

"ToRRGrammar[Regex[...]] is a right regular grammar defined from a " <>

"symbolic regular expression.\n" <>

"ToRRGrammar[Digraph[...]] is a right regular grammar defined from a " <>

"digraph with labeled vertices.";

ToDigraph::usage = "ToDigraph[NFA[...]] is a directed graph defined " <>

"from an NFA.\n" <>

"ToDigraph[DFA[...]] is a directed graph defined from a DFA.\n" <>

"ToDigraph[Regex[...]] is a directed graph defined from a " <>

"symbolic regular expression.\n" <>

"ToDigraph[RRGrammar[...]] is a directed graph defined " <>

"from a right regular grammar.";

RegStar::usage = "RegStar[NFA[...]] performs the star operation on an NFA.\n" <>

"RegStar[DFA[...]] performs the star operation on a DFA.\n" <>

"RegStar[Regex[...]] performs the star operation on a symbolic " <>

"regular expression.\n" <>

"RegStar[RRGrammar[...]] performs the star operation on a right " <>

"regular grammar.\n" <>

"RegStar[Digraph[...]] performs the star operation on a directed graph.";

RegComplement::usage = "RegComplement[NFA[...], {\"str1\", \"str2\",...}] " <>

"performs the " <>

"complement operation on an NFA with respect to a given alphabet.\n" <>

"RegComplement[DFA[...], {\"str1\", \"str2\",...}] performs the complement " <>

"operation on a DFA with respect to a given alphabet.\n" <>

"RegComplement[Regex[...], {\"str1\", \"str2\",...}] performs the " <>

"complement operation on a symbolic regular expression with respect " <>

"to a given alphabet.\n" <>

"RegComplement[RRGrammar[...], {\"str1\", \"str2\",...}] performs the " <>

"complement operation on a right regular grammar with respect to a " <>

"given " <> "alphabet.\n" <>

"RegComplement[Digraph[...], {\"str1\", \"str2\",...}] performs the " <>

"complement " <>

"operation on a directed graph with respect to a given alphabet.";

RegReverse::usage = "RegReverse[NFA[...]] performs the reversal " <>

"operation on an NFA.\n" <>

"RegReverse[DFA[...]] performs the reversal operation on a DFA.\n" <>

"RegReverse[Regex[...]] performs the reversal operation on a symbolic " <>

"regular expression.\n" <>

"RegReverse[RRGrammar[...]] performs the reversal operation on a right " <>

"regular grammar.\n" <>

"RegReverse[Digraph[...]] performs the reversal operation on a directed " <>

"graph.";

RegUnion::usage = "RegUnion[NFA[...], NFA[...]] performs the reversal " <>

"operation on two NFAs.\n" <>

"RegUnion[DFA[...], DFA[...]] performs the reversal operation on two " <>

"DFAs.\n" <>

"RRegUnionRegex[...], Regex[...]] performs the reversal operation on two " <>

"symbolic " <> "regular expressions.\n" <>

"RegUnion[RRGrammar[...], RRGrammar[...]] performs the reversal operation " <>

"on two " <> "right regular grammars.\n" <>

"RegUnion[Digraph[...], Digraph[...]] performs the reversal operation on " <>

"two directed " <> "graphs.";

RegConcat::usage = "RegConcat[NFA[...], NFA[...]] performs the " <>

"concatenation operation on two NFAs.\n" <>

"RegConcat[DFA[...], DFA[...]] performs the concatenation operation on " <>

"two DFAs.\n" <>

"ReRegConcategex[...], Regex[...]] performs the concatenation operation on " <>

"two symbolic " <> "regular expressions.\n" <>

"RegConcat[RRGrammar[...], RRGrammar[...]] performs the concatenation " <>

"operation on two " <> "right regular grammars.\n" <>

"RegConcat[Digraph[...], Digraph[...]] performs the concatenation " <>

"operation on two directed " <> "graphs.";

RegIntersection::usage = "RegIntersection[NFA[...], NFA[...]] performs the " <>

"intersection operation on two NFAs.\n" <>

"RegIntersection[DFA[...], DFA[...]] performs the intersection operation " <>

"on two DFAs.\n" <>

"RegIntersection[Regex[...], Regex[...]] performs the intersection " <>

"operation on two symbolic " <>

"regular expressions.\n" <>

"RegIntersection[RRGrammar[...], RRGrammar[...]] performs the intersection " <>

"operation on two " <> "right regular grammars.\n" <>

"RegIntersection[Digraph[...], Digraph[...]] performs the intersection " <>

"operation on two directed " <> "graphs.";

ToRegularExpression::usage = "ToRegularExpression[Regex[...]] converts a " <>

"symbolic regular " <> "expression to a Mathematica regular expression.";

RegexOr::usage = "RegexOr is a head used to define symbolic regular " <>

"expressions. It " <>

"represents the regular expression equal to the union of its arguments.";

RegexConcat::usage = "RegexConcat is a head used to define symbolic " <>

"regular expressions. It " <>

"represents the regular expression equal to the concatentation of its " <>

"arguments.";

RegexStar::usage = "RegexStar is a head used to define symbolic regular " <>

"expressions. It " <>

"represents the regular expression equal to the star of its argument.";

RRGrammarOr::usage = "RRGrammarOr is a head used to define right regular " <>

"grammars. It " <> "represents a possible choice between its arguments.";

RRGrammarConcat::usage = "RRGrammarConcat is a head used to define right " <>

"regular grammars. It " <>

"is used to concatenate a letter with a nonterminal symbol.";

EmptyWord::usage = "EmptyWord represents the empty word. It may be used in " <>

"symbolic regular expressions.";

GeneratingFunction::usage = "GeneratingFunction[Regex[...], {\"letter1\" -> " <>

"expr1, \"letter2\" -> expr2,...}] " <>

"gives the generating function for the regular language specified by a " <>

"symbolic " <>

"regular expression. It takes as its second argument a list of rules " <>

"giving a marker/weight " <> "for each letter in the alphabet.\n" <>

GeneratingFunction::usage;

Begin["`Private`"] (* Begin Private Context *)

(* all public symbols in this package *)

publicSymbols = {NFA, DFA, Regex, RRGrammar, Digraph,

ToNFA, ToDFA, ToRegex, ToRRGrammar, ToDigraph, ToRegularExpression,

RegStar, RegComplement, RegReverse, RegUnion, RegConcat, RegIntersection,

Disambiguate, AmbiguousQ,

RegexOr, RegexConcat, RegexStar, EmptyWord,

RRGrammarConcat, RRGrammarOr};

(* ::Section:: *)

(* Input validation *)

(* returns a list of the unique nonterminal symbols in a right regular *)

(* grammar *)

nonTerminals[RRGrammar[grammar_]] := Module[

{

extendedGrammar

},

(* expand or[...]'s into separate rules in extendedGrammar *)

expandedGrammar = grammar /. (rhs_ -> RRGrammarOr[args__]) :>

Sequence@@((rhs -> # &) /@ {args});

Union@Flatten[{

grammar[[All, 1]],

Cases[grammar, Verbatim[Rule][lhs_, sym_Symbol /; sym != EmptyWord

] -> {lhs, sym}],

Cases[grammar, Verbatim[Rule][lhs_, sym_Symbol[n_]] -> {lhs, sym[n]}],

Cases[grammar, Verbatim[Rule][lhs_, RRGrammarConcat[_, expr_]] -> expr]

}]

];

NFA::invalid = "Invalid NFA.";

DFA::invalid = "Invalid DFA.";

RegularExpression::invalid = "Invalid Mathematica regular expression.";

Regex::invalid = "Invalid symbolic regular expression.";

RRGrammar::invalid = "Invalid right regular grammar.";

Digraph::invalid = "Invalid directed graph.";

GeneratingFunction::invalidRules = "Invalid rules.";

RegComplement::invalidAlphabet = "Invalid alphabet.";

(* validate NFAs *)

validateRLR[NFA[numStates_Integer?NonNegative, alphabet:{___String},

transitionMatrix_, acceptStates:{___Integer},

initialState:_Integer|Null]] := Module[ {

ok = True

},

(* alphabet is valid *)

ok = ok && !MemberQ[alphabet, ""];

(* matrix entries are valid lists *)

ok = ok && ((numStates == 0 && transitionMatrix == {}) ||

MatrixQ[transitionMatrix, MatchQ[#, {___Integer}] &&

(Max[#] <= numStates && Min[#] >= 1) &]);

(* matrix dimensions are correct *)

ok = ok && ((numStates == 0 && transitionMatrix == {}) ||

Dimensions[transitionMatrix, 2] ==

{numStates, Length[alphabet] + 1});

(* accept states are valid *)

ok = ok && ((numStates == 0 && acceptStates == {}) ||

(numStates > 0 && Max[acceptStates] <= numStates &&

Min[acceptStates] >= 1));

(* initial state is valid *)

ok = ok && ((numStates == 0 && initialState == Null) ||

(numStates > 0 && 1 <= initialState <= numStates));

If[ !ok, Message[NFA::invalid]];

ok

];

validateRLR[NFA[___]] := (Message[NFA::invalid];False);

(* validate DFAs *)

validateRLR[DFA[numStates_Integer?NonNegative, alphabet:{___String},

transitionMatrix_, acceptStates:{___Integer},

initialState:_Integer|Null]] := Module[ {

ok = True

},

(* alphabet is valid *)

ok = ok && !MemberQ[alphabet, ""];

(* matrix entries are valid numbers *)

ok = ok && ((numStates == 0 && transitionMatrix == {}) ||

MatrixQ[transitionMatrix, MatchQ[#, _Integer] &&

(# <= numStates && # >= 1) &]);

(* matrix dimensions are correct *)

ok = ok && ((numStates == 0 && transitionMatrix == {}) ||

Dimensions[transitionMatrix, 2] ==

{numStates, Length[alphabet]});

(* accept states are valid *)

ok = ok && ((numStates == 0 && acceptStates == {}) ||

(numStates > 0 && Max[acceptStates] <= numStates &&

Min[acceptStates] >= 1));

(* initial state is valid *)

ok = ok && ((numStates == 0 && initialState == Null) ||

(numStates > 0 && 1 <= initialState <= numStates));

If[ !ok, Message[DFA::invalid]];

ok

];

validateRLR[DFA[___]] := (Message[DFA::invalid];False);

(* validate Mathematica regular expressions *)

validateRLR[RegularExpression[regex_String]] := Module[

{

ok = True

},

ok = ok && StringMatchQ[regex, (LetterCharacter | DigitCharacter | "\\*" |

"(" |

")" | "|") ...];

(* test validity as a regex according to Mathematica *)

ok = ok && Check[StringMatchQ["", RegularExpression[regex]], $Failed] =!=

$Failed;

If[!ok, Message[RegularExpression::invalid]];

ok

];

validateRLR[RegularExpression[Null]] := True;

validateRLR[RegularExpression[___]] := (Message[RegularExpression::invalid];

False);

(* validate symbolic regex without Regex head *)

validateRawRegex[Null] := True;

validateRawRegex[EmptyWord] := True;

validateRawRegex[str_String /; str != ""] := True;

validateRawRegex[RegexStar[regex_]] := validateRawRegex[regex];

validateRawRegex[RegexOr[regexes__]] := And @@ validateRawRegex /@ {regexes};

validateRawRegex[RegexConcat[regexes__]] := And @@ validateRawRegex /@

{regexes};

validateRawRegex[___] := False;

(* validate symbolic regex *)

validateRLR[Regex[regex_]] := Module[

{

ok = True

},

ok = ok && validateRawRegex[regex];

If[!ok, Message[Regex::invalid]];

ok

];

validateRLR[Regex[___]] := (Message[Regex::invalid];False);

(* validate right regular grammars *)

validateRLR[g:RRGrammar[grammar:{(_ -> _) ...}]] := Module[

{

ok = True,

nonTerms = nonTerminals[g],

validateTerm

},

(* validate left-hand sides *)

ok = ok && MatchQ[nonTerms, {(_Symbol | _Symbol[_Integer]) ...}];

(* validate right-hand sides *)

validateTerm[EmptyWord] := True;

validateTerm[sym_Symbol] := True;

validateTerm[sym_Symbol[n_Integer]] := True;

validateTerm[str_String /; str != ""] := True;

validateTerm[RRGrammarConcat[str_String /; str != "", sym_Symbol]] := True;

validateTerm[RRGrammarConcat[str_String /; str != "", sym_Symbol[

n_Integer]]] := True;

validateTerm[_] := False;

ok = ok && MatchQ[grammar[[All,2]], {( _?validateTerm | RRGrammarOr[

_?validateTerm .. ]) ...}];

If[!ok, Message[RRGrammar::invalid]];

ok

];

validateRLR[RRGrammar[___]] := (Message[RRGrammar::invalid];False);

(* validate digraphs *)

validateRLR[Digraph[graph_Graph, startVertices_List, endVertices_List, True |

False]] := Module[

{

ok = True,

vertices

},

(* validate graph *)

ok = ok && (EmptyGraphQ[graph] || DirectedGraphQ[graph]);

vertices = VertexList[graph];

ok = ok && MatchQ[vertices, {___Integer}];

(* make sure vertex labels are nonempty strings *)

ok = ok && And @@ ( (Head[PropertyValue[{graph, #}, VertexLabels]] ===

String &&

PropertyValue[{graph, #}, VertexLabels] != "") & /@ vertices );

(* validate vertex lists *)

ok = ok && Union[startVertices, endVertices, vertices] == Union[vertices];

If[!ok, Message[Digraph::invalid]];

ok

];

validateRLR[Digraph[___]] := (Message[Digraph::invalid];False);

(* validate rules passed to GeneratingFunction for making letters *)

validateRules[rules_, regex_] := Module[

{

strings = Cases[regex, _String, {0, Infinity}],

ok = True

},

ok = ok && MatchQ[rules, {(_String -> _) ...}];

ok = ok && Complement[strings, rules[[All, 1]]] == {};

If[!ok, Message[GeneratingFunction::invalidRules]];

ok

];

(* validate alphabet passed to RegComplement *)

validateAlphabet[alphabet_] := Module[

{

ok = True

},

ok = ok && MatchQ[alphabet, {_String ...}];

ok = ok && DeleteDuplicates[alphabet] == alphabet;

If[!ok, Message[RegComplement::invalidAlphabet]];

ok

];

(* ::Section:: *)

(* Regex to NFA conversion *)

(* This algorithm is simply described in Flajolet and Sedgewick on p 735 *)

113



regex2nfa[Null] := NFA[0, {}, {}, {}, Null];

regex2nfa[str_String] := NFA[

2,

{str},

{

{{2}, {}},

{{}, {}}

},

{2}, 1

];

regex2nfa[EmptyWord] := NFA[

1, {},

{

{{}}

},

{1}, 1

];

regex2nfa[RegexOr[first_, rest__]] :=

RegUnion[regex2nfa[first], regex2nfa[RegexOr[rest]]];

regex2nfa[RegexOr[regex_]] := regex2nfa[regex];

regex2nfa[RegexStar[regex_]] :=

RegStar[regex2nfa[regex]];

regex2nfa[RegexConcat[first_, rest__]] :=

RegConcat[regex2nfa[first], regex2nfa[RegexConcat[rest]]];

regex2nfa[RegexConcat[regex_]] := regex2nfa[regex];

ToNFA[Regex[regex_], OptionsPattern[]] := Module[

{},

(

regex2nfa[regex]

)/; !OptionValue[validationRequired] || validateRLR[Regex[regex]]

];

(* ::Section:: *)

(* RRGrammar to NFA conversion *)

(* in this straightforward algorithm, the nonterminal symbols become states *)

(* in the NFA *)

ToNFA[RRGrammar[{}], OptionsPattern[]] := NFA[0, {}, {}, {}, Null];

ToNFA[g: RRGrammar[grammar_], OptionsPattern[]] :=

Module[

{

(* added for rhs -> string rules *) acceptState = Unique[],

stateSet,

alphabet = Cases[grammar, _String, Infinity]//Union,

(* lookup functions *) stateNumber, alphabetNumber,

expandedGrammar,

transitionMatrix, acceptStates

},

(

stateSet = Flatten[{nonTerminals[g], acceptState}];

stateNumber[state_ /; MemberQ[stateSet, state]] := Position[stateSet,

state, {1}, Heads -> False][[1,1]];

alphabetNumber[str_] := Position[alphabet,

str, {1}, Heads -> False][[1,1]];

(* initialize acceptStates *)

acceptStates = {stateNumber[acceptState]};

(* expand or[...]'s into separate rules in extendedGrammar *)

expandedGrammar = grammar /. (rhs_ -> RRGrammarOr[args__]) :>

Sequence@@((rhs -> # &) /@ {args});

transitionMatrix = ConstantArray[{}, {Length[stateSet],

Length[alphabet] + 1}];

(* incorporate each rule into transitionMatrix and acceptStates *)

Function[rule,

rule /. {

(lhs_ -> EmptyWord) :> (acceptStates = Union[acceptStates, {

stateNumber[lhs]}];),

(lhs_ -> st:(_Symbol|_Symbol[_])) :> (transitionMatrix[[

stateNumber[lhs], -1]] =

Union[transitionMatrix[[stateNumber[lhs], -1]], {

stateNumber[st]}];),

(lhs_ -> RRGrammarConcat[str_, st:(_Symbol|_Symbol[_])])

:> (transitionMatrix[[stateNumber[lhs],

alphabetNumber[str]]] =

Union[transitionMatrix[[stateNumber[lhs],

alphabetNumber[str]]], {stateNumber[st]}];),

(lhs_ -> str_String)

:> (transitionMatrix[[stateNumber[lhs],

alphabetNumber[str]]] =

Union[transitionMatrix[[stateNumber[lhs],

alphabetNumber[str]]], {stateNumber[acceptState]}];)

}

] /@ expandedGrammar;

NFA[Length[stateSet], alphabet, transitionMatrix, acceptStates,

grammar[[1,1]]//stateNumber]

) /; (!OptionValue[validationRequired] || validateRLR[g])

];

(* ::Section:: *)

(* Digraph to NFA conversion *)

ToNFA[dg: Digraph[graph_, startVertices_, endVertices_, eAccepted_],

OptionsPattern[]] :=

Module[

{

edges = EdgeList[graph],

alphabet =

Union[PropertyValue[{graph, #}, VertexLabels] & /@

VertexList[graph]],

numStates, initialState, acceptState,

transitionMatrix, vertex2LetterPos, edge2StatePos

},

(

(* the states are all edges, plus a start state and an accept state *)

(* lookup alphabet number for vertex *)

vertex2LetterPos[vertex_] :=

Position[alphabet,

PropertyValue[{graph, vertex}, VertexLabels], {1},

Heads -> False][[1,1]];

(* lookup state number for edge *)

edge2StatePos[edge_] :=

Position[edges, edge, {1}, Heads -> False][[1,1]];

{initialState, acceptState, numStates} = Length[edges] + {1, 2, 2};

(* initialize transitionMatrix *)

transitionMatrix =

ConstantArray[{}, {numStates, Length[alphabet] + 1}];

(* add emove from initial state to accept state if eAccepted *)

If[eAccepted,

transitionMatrix[[initialState, -1]] = {acceptState}];

(* if a vertex is a start vertex and an end vertex, add an

edge labeled with its label from initialState to acceptState *)

Map[Function[vertex,

transitionMatrix[[initialState, vertex2LetterPos[vertex]]] =

Union[transitionMatrix[[initialState,

vertex2LetterPos[vertex]]], {acceptState}];],

Intersection[endVertices, startVertices]];

(* if an edge begins at a start vertex, add an edge labeled with

that start vertex's label between initialState and the state

corresponding to the edge in the digraph *)

Map[Function[edge,

transitionMatrix[[initialState, vertex2LetterPos[edge[[1]]]]] =

Union[

transitionMatrix[[initialState,

vertex2LetterPos[edge[[1]]]]], {edge2StatePos[edge]}];],

EdgeList[graph, DirectedEdge[Alternatives@@startVertices, _]]

];

(* if an edge ends at an end state, add a transition from that edge

to acceptState labeled by the end state's label *)

Map[Function[edge,

transitionMatrix[[edge2StatePos[edge],

vertex2LetterPos[edge[[2]]]]] =

Union[transitionMatrix[[edge2StatePos[edge],

vertex2LetterPos[edge[[2]]]]], {acceptState}];],

EdgeList[graph, DirectedEdge[_, Alternatives@@endVertices]]

];

(* if one edge goes into a vertex and one edge comes out,

add a transition between the edges labeled with the label

of the common vertex *)

Map[Function[edgePair,

transitionMatrix[[edge2StatePos[edgePair[[1]]],

vertex2LetterPos[edgePair[[1, 2]]]]] =

Union[transitionMatrix[[edge2StatePos[edgePair[[1]]],

vertex2LetterPos[edgePair[[1, 2]]]]], {edge2StatePos[

edgePair[[2]]]}];],

Union@Flatten[

Outer[List, EdgeList[graph, DirectedEdge[_, #]],

EdgeList[graph, DirectedEdge[#, _]]] & /@ VertexList[graph],

2]

];

NFA[numStates, alphabet, transitionMatrix, {acceptState},

initialState]

)/; !OptionValue[validationRequired] || validateRLR[dg]

];

(* ::Section:: *)

(* conversion from NFA to DFA (determinization) *)

(* hopcroft implements the algorithm at http://bit.ly/1kT6P2n directly *)

hopcroft[DFA[numStates:(0|1), alphabet_, transitionMatrix_, acceptStates_,

initialState_]] :=

DFA[numStates, alphabet, transitionMatrix, acceptStates, initialState];

hopcroft[DFA[numStates_, alphabet_, transitionMatrix_, acceptStates_,

initialState_]] := Module[

{

p = {Sort@acceptStates,

Complement[Range[numStates], acceptStates]},

w = {Sort@acceptStates}, a, c, x

},

While[Length[w] > 0,

a = First[w]; w = Delete[w, 1];

For[c = 1, c <= Length[alphabet], c++,

x = Flatten@

Position[transitionMatrix[[All, c]],

_?(MemberQ[a, #] &), {1}, Heads -> False];

Map[

(

p =

Union[p /. # :>

Sequence[Intersection[x, #], Complement[#, x]]];

If[MemberQ[w, #],

w =

Union[w /. # :>

Sequence[Intersection[x, #], Complement[#, x]]],

(* else *)

If[Length[Intersection[x, #]] <= Length[Complement[#, x]],

w = Union[w, {Intersection[x, #]}],

(* else *)

w = Union[w, {Complement[#, x]}]

(* end else *)

]

(* end else *)

]

) &,

Select[p, Intersection[#, x] != {} &]

];

];

];

p = DeleteCases[p, {}];

DFA[

Length[p],

alphabet,

Table[

First@Flatten@

Position[

p, _?(MemberQ[#, transitionMatrix[[First@p[[s]], c]]] &), {1},

Heads -> False], {s, 1, Length[p]}, {c, 1, Length[alphabet]}],

Flatten@

Position[p, _?(Intersection[#, acceptStates] != {} &), {1},

Heads -> False],

First@

Flatten@Position[p, _?(MemberQ[#, initialState] &), {1},

Heads -> False]

]

];

SetAttributes[floydWarshall, HoldFirst];

(* this computes transitive closure of a graph via the Floyd-Warshall *)

(* algorithm *)

(* "pass by reference": assigns to the symbol passed *)

floydWarshall[m_] := Module[

{n = Length[m],i,j,k},

For[k = 1, k <= n, ++k,

For[i = 1, i <= n, ++i,

For[j = 1, j <= n, ++j,

m[[i, j]] = m[[i, j]] || (m[[i, k]] && m[[k, j]]);

]

]

];

];

(* deadStates finds states with no path to an accept state in an NFA *)

deadStates[NFA[numStates_, _, transitionMatrix_, acceptStates_, _]] :=

Module[

{adj = Flatten /@ transitionMatrix},

graph = Graph[

Range[numStates],

MapIndexed[DirectedEdge[First[#2], #1] &, adj, {2}] // Flatten // Union];

Complement[Range[numStates],

VertexInComponent[graph, acceptStates]

]

];

ToDFA[nfa:NFA[0, alphabet_, _, _, _], OptionsPattern[]] := Module[

{},

(

DFA[0, alphabet, {}, {}, Null]

) /; !OptionValue[validationRequired] || validateRLR[nfa]

];

ToDFA[nfa :

NFA[numStates_, alphabet_, transitionMatrix_, acceptStates_,

initialState_], OptionsPattern[]] := Module[

{

states = {}, queue,

adjacency =

Table[ReplacePart[ConstantArray[False, numStates],

List /@ (transitionMatrix[[k, -1]]) -> True], {k, 1,

numStates}],

nfaDeadStates = deadStates[nfa],

(* function that stores transitions in the DFA *) tr,

state, neighbor,

dfaInitialState, dfaAcceptStates, stateNumber,

dfaTransitionMatrix

},

(

If[MemberQ[nfaDeadStates, initialState],

Return[DFA[0, {}, {}, {}, Null]]];

floydWarshall[adjacency];

(* incrementally search for possible states in the DFA *)

queue = {Complement[

Flatten[{initialState,

Position[adjacency[[initialState]], True]}], nfaDeadStates]};

While[Length[queue] > 0,

(* process "state": the next state that was found *)

{state, queue} = {First[queue], Rest[queue]};

If[! MemberQ[states, state],

For[i = 1, i <= Length[alphabet], ++i,

(* assign to "neighbor" the set of accessible states from "state" *)

neighbor =

Flatten[Map[

Function[

st, {transitionMatrix[[st, i]],

Map[Position[adjacency[[#]], True] &,

transitionMatrix[[st, i]]]}], state

]]~Complement~nfaDeadStates; (* remove dead states from list *)

(* store neighbor in tr *)

tr[state, i] = neighbor;

queue = Append[queue, neighbor];

];

states = Append[states, state];

];

];

tr[_, _] = {};

stateNumber[set_] :=

Position[states, set, {1}, Heads -> False][[1, 1]];

dfaAcceptStates =

stateNumber /@

Select[states, (Intersection[#, acceptStates] != {}) &];

dfaTransitionMatrix =

Table[stateNumber@tr[states[[s]], a], {s, 1, Length[states]}, {a, 1,

Length[alphabet]}];

DFA[Length[states], alphabet, dfaTransitionMatrix,

dfaAcceptStates, 1] // hopcroft (* minimize just before returning *)

) /; ! OptionValue[validationRequired] || validateRLR[nfa]

];

(* ::Section:: *)

(* conversion from DFA to Regex *)

(* simplifyRawRegex takes a symbolic regex, without the head Regex, and *)

(* simplifies it if possible by removing redundancies *)

simplifyRawRegex[regex_] := FixedPoint[

Replace[#, {

RegexStar[RegexStar[args__]] :> RegexStar[args],

RegexStar[Null] -> EmptyWord,

RegexStar[EmptyWord] -> EmptyWord,

RegexOr[x_] :> x,

RegexOr[i___, Null, f___] :> RegexOr[i, f],

RegexOr[i___, x_, m___, x_, f___] :> RegexOr[i, x, m, f],

RegexOr[i___, RegexOr[args__], f___] :> RegexOr[i, args, f],

RegexConcat[i___, RegexConcat[args__], f___] :> RegexConcat[i, args, f],

RegexConcat[x_] :> x,

RegexConcat[___, Null, ___] -> Null,

RegexConcat[i___, EmptyWord, f___] :> RegexConcat[i, f]

}, {0, Infinity}] &,

regex]

ToRegex[DFA[0, _, _, _, _], OptionsPattern[]] := Regex[Null];

ToRegex[dfa: DFA[numStates_, alphabet_, transitionMatrix_, acceptStates_,

initialState_], OptionsPattern[]] :=

Module[

{

(* edgeRawRegex is a function that stores the current regex corresponding *)

(* to each transition during the conversion process *)

edgeRawRegex,

k, i, j},

(

edgeRawRegex[_, _] = Null;

MapIndexed[

Function[{endState, pos},

(*pos[[1]]=startState, pos[[2]]=letter*)

edgeRawRegex[pos[[1]], endState] =

RegexOr[edgeRawRegex[pos[[1]], endState], alphabet[[pos[[2]]]]];],

transitionMatrix, {2}];

(* setup beginning and end states *)

edgeRawRegex[0, initialState] = EmptyWord;

edgeRawRegex[x_?(MemberQ[acceptStates, #] &), numStates + 1] =

EmptyWord;

For[k = 1, k <= numStates, ++k,

Do[edgeRawRegex[i, j] =

RegexOr[edgeRawRegex[i, j],

RegexConcat[edgeRawRegex[i, k], RegexStar[edgeRawRegex[k, k]],

edgeRawRegex[k, j]]] // simplifyRawRegex, {i,

Append[Range[k + 1, numStates + 1], 0]}, {j,

Append[Range[k + 1, numStates + 1], 0]}

];

];

Regex[edgeRawRegex[0, numStates + 1]]

)/; !OptionValue[validationRequired] || validateRLR[dfa]

];

(* ::Section:: *)

(* conversion from NFA to RRGrammar *)

ToRRGrammar[NFA[0, _, _, _, _], OptionsPattern[]] := RRGrammar[{}];

(* simply convert states to nonterminal symbols *)

ToRRGrammar[nfa: NFA[numStates_, alphabet_, transitionMatrix_, acceptStates_,

initialState_], OptionsPattern[]] :=

Module[

{

nonTermHead = Unique[],

nonTerms,

augmentedAlphabet = {alphabet, EmptyWord}//Flatten,

grammar,

posOfInitial

},

(

(* create nonterminal for each state *)

nonTerms = nonTermHead /@ Range[numStates];

(* convert transitions to grammar productions *)

grammar = Flatten@MapIndexed[

(* #1 == list of stateNums *)

(* #2[[1]] == stateNum, #2[[2]] == letterNum *)

(nonTermHead[#2[[1]]] -> RRGrammarOr @@ (Function[st,

RRGrammarConcat[augmentedAlphabet[[#2[[2]]]],

nonTermHead[st]]] /@ #1)) &,

transitionMatrix, {2}

];

(* simplify resulting grammar *)

grammar = grammar /. RRGrammarConcat[EmptyWord, x_] :> x;

grammar = grammar /. (_ -> RRGrammarOr[]) :> Sequence[];

grammar = grammar /. (lhs_ -> RRGrammarOr[arg_]) :> (lhs -> arg);

(* handle acceptStates *)

grammar = {grammar, (nonTermHead[#] -> EmptyWord) & /@ acceptStates}//

Flatten;

(* if the initial state has no outgoing edges,

and is not an accepting state, return {} *)

If[FreeQ[grammar[[All, 1]], nonTermHead[initialState]], Return[

RRGrammar[{}]]];

(* if start state appears, make one of its rules the first one *)

posOfInitial = Position[grammar, nonTermHead[initialState]][[1, 1]];

If[posOfInitial != 1,

grammar = Permute[grammar, Cycles[{{1, posOfInitial}}]]

];

RRGrammar[grammar]

)/; !OptionValue[validationRequired] || validateRLR[nfa]

];

(* ::Section:: *)

(* conversion from DFA to Digraph *)

cartesian[lists___List] := Flatten[Outer[List, lists, 1], Length[{lists}] - 1];

ToDigraph[DFA[0, _, _, _, _], OptionsPattern[]] := Digraph[Graph[{}], {}, {},

False];

ToDigraph[dfa: DFA[numStates_, alphabet_, transitionMatrix_, acceptStates_,

initialState_], OptionsPattern[]] := Module[

{

(* each transition corresponds to a vertex *)

vertexSet = Flatten[MapIndexed[

(* #1 = to, #2[[1]] = from, #2[[2]] = letterNum *)

{#2[[1]], #1, #2[[2]]}&, transitionMatrix, {2}], 1],

vertexNum,

edgeSet,

graph,

startVertices, endVertices, eAccepted

},

(

(* ranking functon *)

vertexNum[vertex_] := Position[vertexSet, vertex,

{1}, Heads -> False][[1, 1]];

startVertices = vertexNum /@ Cases[vertexSet, {initialState, _, _}];

endVertices = vertexNum /@ Cases[vertexSet, {_, st_, _} /;

MemberQ[acceptStates, st]];

eAccepted = MemberQ[acceptStates, initialState];

(* create the edge set *)

edgeSet = Cases[cartesian[vertexSet, vertexSet],

{{_, c_, _}, {c_, _, _}}];

edgeSet = Map[vertexNum, edgeSet, {2}];

edgeSet = edgeSet /. {i_, f_} :> DirectedEdge[i, f];

graph = Graph[Range[Length[vertexSet]], edgeSet];

(* start vertices are colored green, end vertices have dotted edges *)

(PropertyValue[{graph, #}, VertexStyle] = Green;)& /@ startVertices;

(PropertyValue[{graph, #}, VertexStyle] = EdgeForm[{Thick, Dotted}];

)& /@ endVertices;

(PropertyValue[{graph, #}, VertexStyle] =

Directive[Green, EdgeForm[{Thick, Dotted}]];)& /@ Intersection[

startVertices, endVertices];

(* add String labels to the vertices *)

(# /. {from_, to_, letterNum_} :> (PropertyValue[{graph,

vertexNum[{from, to, letterNum}]}, VertexLabels] =

alphabet[[letterNum]];))& /@ vertexSet;

Digraph[graph, startVertices, endVertices, eAccepted]

)/; !OptionValue[validationRequired] || validateRLR[dfa]

];

(* ::Section:: *)

(* Regular expression conversions *)

(* conversion from Mathematica regular expression to symbolic regular *)

(* expression *)

(* "," not allowed in the regex *)

(* pars takes a Mathematica regular expression and comes up with a symbolic *)

(* regex (except for the head Regex) *)

pars[""] := EmptyWord;

pars[regex_] :=

Module[{temp=regex},

(*insert commas between things that are concatenated*)

temp = FixedPoint[

StringReplace[#,

a_ ~~ b_ /;

a != "(" && b != ")" && b != "*" && a != "," && b != "," &&

a != "|" && b != "|" :> a <> "," <> b] &, temp];

(*insert EmptyWord*)

temp = FixedPoint[

StringReplace[#,

{

"||" -> "|EmptyWord|",

"(|" -> "(EmptyWord|",

"|)" -> "|EmptyWord)",

"()" -> "(EmptyWord)"

}] &, temp];

(*make the star operation the squaring operation *)

temp = StringReplace[temp, "*" -> "^2"];

(*make the concatenation operation the ** operation *)

temp = StringReplace[temp, "," -> "**"];

(* use ToExpression to obtain a Mathematica expression from the string *)

temp = ToExpression[temp, InputForm, Hold];

(* replace the operations present in the expressions with the ones *)

(* found in Regex expressions *)

temp =

Replace[temp,

sym_Symbol?(

!MemberQ[{Hold, RegexOr, RegexConcat, RegexStar,

EmptyWord}, #] &) :>

Module[{chars = Characters@ToString[Unevaluated[sym]]},

If[Length[chars] == 1, chars[[1]],

RegexConcat @@ chars

]

], Infinity];

temp =

FixedPoint[

Replace[#, {Power[expr_, _] :> RegexStar[expr],

NonCommutativeMultiply[args__] :> RegexConcat[args],

Verbatim[Alternatives][args__] :> RegexOr[args]}, {0,

Infinity}] &, temp];

ReleaseHold[temp]

];

ToRegex[RegularExpression[Null], OptionsPattern[]] := Regex[Null];

ToRegex[r:RegularExpression[regex_], OptionsPattern[]] := Module[

{},

Regex@pars["(" <> regex <> ")"] /; !OptionValue[validationRequired] ||

validateRLR[r]

];

(* symbolic regular expression to Mathematica regular expression conversion *)

regex2regularexpression[Null] := Null;

regex2regularexpression[str_String] := str;

regex2regularexpression[EmptyWord] := "()";

regex2regularexpression[RegexStar[expr_]] := "(" <>

regex2regularexpression[expr] <> ")*";

regex2regularexpression[RegexConcat[args__]] := StringJoin[

regex2regularexpression /@ {args}];

regex2regularexpression[RegexOr[args__]] := "(" <>

StringJoin[Riffle[regex2regularexpression /@ {args},"|"]] <> ")";

ToRegularExpression[r:Regex[regex_], OptionsPattern[]] := Module[

{simp = simplifyRawRegex[regex]},

RegularExpression[regex2regularexpression[simp]] /;

!OptionValue[validationRequired] || validateRLR[r]

];

(* ::Section:: *)

(* Closure properties *)

(* Kleene star for right regular grammars *)

RegStar[RRGrammar[{}], OptionsPattern[]] := RRGrammar[{Unique[] -> EmptyWord}];

RegStar[grammar : RRGrammar[_], OptionsPattern[]] := Module[

{

initial1 = grammar[[1, 1, 1]]

},

(

RRGrammar[Flatten[{ initial1 -> EmptyWord,

Replace[First[grammar],

{

Verbatim[Rule][nonTerm_, EmptyWord] :> (nonTerm ->

RRGrammarOr[initial1, EmptyWord]),

RRGrammarOr[pre___, EmptyWord, post___] :>

RRGrammarOr[pre, EmptyWord, initial1, post]

},

Infinity

]}]]

)/; !OptionValue[validationRequired] || validateRLR[grammar]

];

(* complement for DFAs *)

RegComplement[dfa:DFA[numStates_, alphabet_, transitionMatrix_, acceptStates_,

initialState_], compAlphabet_, OptionsPattern[]] := Module[

{

compDFA,

intersectionAlphabet = Intersection[alphabet, compAlphabet],

intersectionAlphabetRegex,

compNotAlphabetLetters = Complement[compAlphabet, alphabet],

compNotAlphabetRegex

},

(

(* compute the DFA that accepts the words, over the alphabet "alphabet", *)

(* that dfa rejects *)

compDFA = If[numStates == 0,

(* no states *)

If[alphabet == {},

ToDFA[Regex[EmptyWord]],

ToDFA[Regex[RegexStar[RegexOr@@alphabet]]]

],

(* states *)

DFA[numStates, alphabet, transitionMatrix,

Complement[Range[numStates], acceptStates], initialState]

];

(* define a regex that parses all words with letters found in alphabet *)

(* and compAlphabet *)

intersectionAlphabetRegex = If[intersectionAlphabet == {},

Regex[EmptyWord],

Regex[RegexStar[RegexOr@@intersectionAlphabet]]

];

(* define a regex that parses all words containg a letter in compAlphabet *)

(* but not alphabet *)

compNotAlphabetRegex = If[compNotAlphabetLetters == {},

Regex[Null],

Regex[RegexConcat[

RegexStar[RegexOr@@compAlphabet],

RegexOr@@compNotAlphabetLetters,

RegexStar[RegexOr@@compAlphabet]]

]

];

(* return the DFA that accepts any words with a letter in compAlphabet *)

(* that is not in alphabet AND any words over compAlphabet that dfa *)

(* rejects *)

RegUnion[RegIntersection[ToRegex[compDFA], intersectionAlphabetRegex],

compNotAlphabetRegex]//ToDFA

)/; !OptionValue[validationRequired] ||

(validateRLR[dfa] && validateAlphabet[compAlphabet])

];

(* reverse for regexes *)

(* regexReverse reverses a symbolic regular expression lacking the head *)

(* Regex *)

regexReverse[str_String] := str;

regexReverse[Null] := Null;

regexReverse[RegexOr[args__]] := RegexOr @@ regexReverse /@ {args};

regexReverse[EmptyWord] := EmptyWord;

regexReverse[RegexConcat[args__]] :=

RegexConcat @@ regexReverse /@ Reverse[{args}];

regexReverse[RegexStar[arg_]] := RegexStar[regexReverse[arg]];

RegReverse[Regex[regex_], OptionsPattern[]] := Module[

{},

(

Regex[regexReverse[regex]]

)/; !OptionValue[validationRequired] || validateRLR[Regex[regex]]

];

(* union for right regular grammars *)

RegUnion[grammar1 : RRGrammar[_], RRGrammar[{}], OptionsPattern[]] := Module[

{},

(

grammar1

)/; !OptionValue[validationRequired] || validateRLR[grammar1]

];

RegUnion[RRGrammar[{}], grammar2 : RRGrammar[_], OptionsPattern[]] := Module[

{},

(

grammar2

)/; !OptionValue[validationRequired] || validateRLR[grammar2]

];

RegUnion[grammar1 : RRGrammar[_], grammar2 : RRGrammar[_], OptionsPattern[]] :=

Module[

{

nonTerms1 = nonTerminals[grammar1],

nonTerms2 = nonTerminals[grammar2],

initial1, initial2,

commonNonTerms,

replacements, rules

},

(

(* do some stuff to make sure the two grammars don't share any *)

(* nonterminals *)

commonNonTerms = Intersection[nonTerms1, nonTerms2];

replacements = Table[Unique[], {Length[commonNonTerms]}];

rules = MapThread[Rule, {commonNonTerms, replacements}];

initial1 = grammar1[[1, 1, 1]];

initial2 = grammar2[[1, 1, 1]] /. rules;

(* combine the grammars *)

RRGrammar[{Unique[] -> RRGrammarOr[initial1, initial2],

grammar1//First, grammar2 /. rules //First}//Flatten]

)/; !OptionValue[validationRequired] ||

(validateRLR[grammar1] && validateRLR[grammar2])

];

(* concatenation for right regular grammars *)

RegConcat[grammar1 : RRGrammar[_], RRGrammar[{}], OptionsPattern[]] := Module[

{},

RRGrammar[{}] /; !OptionValue[validationRequired] ||

validateRLR[grammar1]

];

RegConcat[grammar1 : RRGrammar[_], grammar2 : RRGrammar[_], OptionsPattern[]] :=

Module[

{

nonTerms1 = nonTerminals[grammar1],

nonTerms2 = nonTerminals[grammar2],

initial2,

commonNonTerms,

replacements, rules

},

(

(* do some stuff to make sure the two grammars don't share any *)

(* nonterminals *)

commonNonTerms = Intersection[nonTerms1, nonTerms2];

replacements = Table[Unique[], {Length[commonNonTerms]}];

rules = MapThread[Rule, {commonNonTerms, replacements}];

initial2 = grammar2[[1, 1, 1]] /. rules;

RRGrammar[Flatten[{

First[grammar1] /. {

Verbatim[Rule][nonTerm_, EmptyWord] :>

(nonTerm -> initial2),

RRGrammarOr[pre___, EmptyWord, post___] :>

RRGrammarOr[pre, initial2, post]

}, First[grammar2] /. rules

}]]

)/; !OptionValue[validationRequired] ||

(validateRLR[grammar1] && validateRLR[grammar2])

];

(* intersection for DFAs *)

RegIntersection[dfa1:DFA[numStates1_, alphabet1_, transitionMatrix1_,

acceptStates1_, initialState1_], dfa2:DFA[numStates2_, alphabet2_,

transitionMatrix2_,

acceptStates2_, initialState2_], OptionsPattern[]] := Module[

{

newAlphabet = Intersection[alphabet1, alphabet2],

(* new state set is cartesian product of state sets *)

stateSet = cartesian[Range[numStates1], Range[numStates2]],

stateNum,

newTransitionMatrix, newAcceptStates, newInitialState,

alphabet1Num, alphabet2Num

},

(

If[numStates1 == 0 || numStates2 == 0,

Return[DFA[0, newAlphabet, {}, {}, Null]];

];

(* the alphabet1Num function looks up the position of a letter in *)

(* alphabet1 *)

alphabet1Num[alphabetNum_] := Position[alphabet1,

newAlphabet[[alphabetNum]], {1}, Heads -> False][[1, 1]];

(* the alphabet2Num function looks up the position of a letter in *)

(* alphabet2 *)

alphabet2Num[alphabetNum_] := Position[alphabet2,

newAlphabet[[alphabetNum]], {1}, Heads -> False][[1, 1]];

(* the stateNum function looks up the position of a state in stateSet *)

stateNum[state_] := Position[stateSet, state, {1}, Heads-> False][[1, 1]];

newInitialState = stateNum[{initialState1, initialState2}];

newAcceptStates = stateNum /@ Select[stateSet,

MemberQ[acceptStates1, #[[1]]] && MemberQ[acceptStates2, #[[2]]]&];

(* compute new transition matrix *)

newTransitionMatrix = Table[

stateNum[

{

transitionMatrix1[[stateSet[[state]][[1]], alphabet1Num[c]]],

transitionMatrix2[[stateSet[[state]][[2]], alphabet2Num[c]]]

}

],

{state, 1, Length[stateSet]},

{c, 1, Length[newAlphabet]}

];

DFA[Length[stateSet], newAlphabet, newTransitionMatrix,

newAcceptStates, newInitialState]

)/; !OptionValue[validationRequired] ||

(validateRLR[dfa1] && validateRLR[dfa2])

];

(* ::Section:: *)

(* To generating functions *)

(* takes an unambiguous regex lacking the head Regex and returns the *)

(* corresponding generating function *)

regex2GF[regex_, rules_] :=

FixedPoint[Replace[#,

{

str_String :> (str /. rules),

Null -> 0,

EmptyWord -> 1,

RegexOr[args__] :> Plus[args],

RegexConcat[args__] :> Times[args],

RegexStar[arg_] :> 1/(1 - arg)

},

{0, Infinity}]&,

regex

];

previousOptions = Options[GeneratingFunction]

Unprotect[GeneratingFunction];

Options[GeneratingFunction] = Union[

previousOptions,

{validationRequired -> True}

];

Protect[GeneratingFunction];

Unprotect[GeneratingFunction];

GeneratingFunction[regex:Regex[_], rules_, OptionsPattern[]] :=

Module[

{},

(

(* disambiguate the regex and pass to regex2GF *)

regex2GF[First@ToRegex@ToDFA@regex, rules]

)/; !OptionValue[validationRequired] || validateRules[rules, regex]

];

Protect[GeneratingFunction];

(* ::Section:: *)

(* Code generation for repetitive bits and pieces *)

(* in this section we do dynamic code generation to set up the automatic *)

(* conversion necessary before and/or after the operations and conversions *)

(* that are implemented above *)

(* for example, we implemented the intersection operation for DFAs only; this *)

(* section defines how to convert each RLR to and from DFAs in order to *)

(* use that implementation *)

viaAlgorithms = {

(* {to, from, via} *)

{ToNFA, DFA, ToRegex},

{ToDFA, Regex, ToNFA},

{ToDFA, RRGrammar, ToNFA},

{ToDFA, Digraph, ToNFA},

{ToRegex, NFA, ToDFA},

{ToRegex, RRGrammar, ToDFA},

{ToRegex, Digraph, ToDFA},

142



{ToRRGrammar, DFA, ToNFA},

{ToRRGrammar, Regex, ToNFA},

{ToRRGrammar, Digraph, ToNFA},

{ToDigraph, NFA, ToDFA},

{ToDigraph, Regex, ToDFA},

{ToDigraph, RRGrammar, ToDFA}

};

(* define conversion algorithms that go via

another conversion *)

(# /. {to_, from_, via_} :>

Hold[

to[name : from[___], opts:OptionsPattern[]] :=

to[via[name, opts], validationRequired -> False];

]) & /@

viaAlgorithms // ReleaseHold;

(* complement *)

complementVias = {NFA, Regex, RRGrammar, Digraph};

(Hold[

RegComplement[name : #[___], alpha_, opts:OptionsPattern[]] :=

ToExpression["To" <> ToString[#]][

RegComplement[ToDFA[name, opts], alpha, validationRequired -> False]

];

]) & /@

complementVias // ReleaseHold;

(* star and reverse *)

starReverseVias = {

{RegStar, DFA, ToRRGrammar},

{RegStar, Regex, ToRRGrammar},

{RegStar, NFA, ToRRGrammar},

{RegStar, Digraph, ToRRGrammar},

{RegReverse, NFA, ToRegex},

{RegReverse, DFA, ToRegex},

{RegReverse, RRGrammar, ToRegex},

{RegReverse, Digraph, ToRegex}

};

(# /. {op_, rlr_, via_} :> Hold[

op[name : rlr[___], opts:OptionsPattern[]] :=

ToExpression["To" <> ToString[rlr]][

op[via[name, opts], validationRequired -> False]

];

]) & /@

starReverseVias // ReleaseHold;

(* union, concat, intersection *)

uciVias = {

{RegUnion, DFA, ToRRGrammar},

{RegUnion, Regex, ToRRGrammar},

{RegUnion, NFA, ToRRGrammar},

{RegUnion, Digraph, ToRRGrammar},

{RegConcat, DFA, ToRRGrammar},

{RegConcat, Regex, ToRRGrammar},

{RegConcat, NFA, ToRRGrammar},

{RegConcat, Digraph, ToRRGrammar},

{RegIntersection, NFA, ToDFA},

{RegIntersection, Regex, ToDFA},

{RegIntersection, RRGrammar, ToDFA},

{RegIntersection, Digraph, ToDFA}

};

(# /. {op_, rlr_, via_} :> Hold[

op[name1 : rlr[___], name2 : rlr[___], opts:OptionsPattern[]] :=

ToExpression["To" <> ToString[rlr]][

op[via[name1, opts], via[name2, opts], validationRequired -> False]

];

]) & /@

uciVias // ReleaseHold;

(* add options for downvalued symbols except GeneratingFunction *)

(* also the symbols with catch-all definitions *)

symbolsWithOptions = {

ToNFA, ToDFA, ToRegex, ToRRGrammar, ToDigraph, ToRegularExpression,

RegStar, RegComplement, RegReverse, RegUnion, RegConcat, RegIntersection

};

(MessageName[#, "invalidArgumentSyntax"] = "Invalid argument syntax.")& /@

symbolsWithOptions;

Hold[#[___] /; (Message[#::invalidArgumentSyntax]; False) := Null;]& /@

symbolsWithOptions //ReleaseHold;

(Options[#] = {validationRequired -> True})& /@ symbolsWithOptions;

End[] (* End Private Context *)

EndPackage[]

GFeqn2coefs.m

(* Mathematica Package *)

BeginPackage["Genfunlib`GFeqn2coefs`"]

CoefsByDerivs::usage = "CoefsByDerivs[{eqn1, eqn2,...}, {f,g, ...}, " <>

"{x, 0, nx}, {y, 0, ny}, ...] gives the Taylor expansions at (0,0,...) " <>

"of the power series f,g,... satisfying the equations.";

CoefsByNewton::usage = "CoefsByNewton[eqn, f, {x, 0, nx}] gives the " <>

"Maclaurin expansion in x of f satisfying eqn.";

Begin["`Private`"] (* Begin Private Context *)

(* what can usually be used as a variable, according to

ref/message/General/ivar *)

variablePattern = Except[_String | _?NumberQ | _Plus | _Times |

_Sum | _Product | _^_Integer];

CoefsByDerivs::invalid = "Invalid input.";

CoefsByNewton::invalid = "Invalid input.";

CoefsByDerivs::invalidArgumentSyntax = "Invalid argument syntax.";

CoefsByNewton::invalidArgumentSyntax = "Invalid argument syntax.";

(* validate the equation or system of equations that the user supplies *)

validateSystem[system : {HoldPattern[_ == _]..}, series: { variablePattern.. },

iters:({variablePattern, 0, _Integer?NonNegative}..)] := Module[

{

ok = True

},

If[ !ok, Message[CoefsByDerivs::invalid]];

ok

];

validateSystem[___] := (Message[CoefsByDerivs::invalid];False);

validateEquation[lhs_ == rhs_, series:variablePattern,

{var:variablePattern, 0, _Integer?NonNegative}] := Module[

{ok = True},

ok = ok && !FreeQ[series, var] && !FreeQ[lhs-rhs, series];

ok = ok && (D[lhs - rhs, series] /. {var->0, series->0}) =!= 0;

If[ !ok, Message[CoefsByNewton::invalid]];

ok

];

validateEquation[___] := (Message[CoefsByNewton::invalid];False);

CoefsByDerivs[x:Except[_List], y__] := CoefsByDerivs[{x}, y];

CoefsByDerivs[x_, y:Except[_List], z__] := CoefsByDerivs[x, {y}, z];

CoefsByDerivs[system_, series_, iters__] := Module[

{

indets = {iters} /. {x_, 0, _} :> x,

systemCanon = system /. (l_ == r_ :> l - r),

expansion,

eqns, subs

},

(

expansion = Normal[Series[systemCanon, iters]];

eqns = Thread[Flatten[CoefficientList[expansion, indets]] == 0];

subs = Flatten[Solve[eqns]];

Thread[series == (Series[series, iters] /. subs)]

) /; validateSystem[system, series, iters]

];

CoefsByDerivs[___] /; (Message[CoefsByDerivs::invalidArgumentSyntax]; False) :=

Null;

CoefsByNewton[lhs_ == rhs_, series_, {var_, 0, n_}] := Module[

{

f = lhs - rhs, ff, m = 2^Ceiling[Log[2, n]],

approx = First[

(series/.var->0) /. Solve[(lhs==rhs)/.var->0, series/.var->0]]

},

(

ff[z_] := (f /. series :> z);

Do[

approx = Normal[# - ff[#]/ff'[#] &[approx] + O[var]^(2^k)],

{k, 1, Log[2, m] + 1}

];

approx + O[var]^(n + 1)

)/; validateEquation[lhs==rhs, series, {var, 0, n}]

];

CoefsByNewton[___] /; (Message[CoefsByNewton::invalidArgumentSyntax]; False) :=

Null;

End[] (* End Private Context *)

EndPackage[]

rec2GFeqn.m

(* Mathematica Package *)

BeginPackage["Genfunlib`rec2GFeqn`"]

(* this file defines no public functions, however it does *)

(* override GeneratingFunction*)

Begin["`Private`"]

(* http://bit.ly/1lgtAt1 *)

GeneratingFunction;

protected = Unprotect[GeneratingFunction];

(* what can usually be used as a variable, according to

ref/message/General/ivar *)

variablePattern = Except[_String | _?NumberQ | _Plus | _Times |

_Sum | _Product | _^_Integer];

arg2 = Sequence[n:variablePattern, x:variablePattern];

(* rational function (over syntactic integers) coefficients *)

(* http://mathematica.stackexchange.com/a/32611/208 *)

(* partial fraction expansion *)

GeneratingFunction[expr_, arg2,

149



opts:OptionsPattern[]] /; expr =!= Apart[expr, n] :=

GeneratingFunction[Apart[expr, n], n, x, opts];

GeneratingFunction[expr_ * (n_ + k_Integer?Positive)^(j_Integer?Negative), arg2,

opts:OptionsPattern[]] := With[

{t = Unique[]},

x^(-k) * Integrate[t^(k-1) * GeneratingFunction[expr * (n + k)^(j + 1),

n, t, opts], {t, 0, x}]

];

(* symbolic sums *)

GeneratingFunction[Sum[expr_, {i_, lb_, ub_}], arg2, opts:OptionsPattern[]] /;

FreeQ[lb, n] && FreeQ[ub, n] := Sum[GeneratingFunction[expr, n, x, opts],

{i, lb, ub}];

(* different ways to set terms to 0 *)

GeneratingFunction[Boole[expr_] * fac_, arg2, opts:OptionsPattern[]] :=

GeneratingFunction[Piecewise[{{fac, expr}}], n, x, opts];

GeneratingFunction[UnitStep[expr_] * fac_, arg2, opts:OptionsPattern[]] :=

GeneratingFunction[Piecewise[{{fac, expr >= 0}}], n, x, opts];

GeneratingFunction[Piecewise[{{expr_, Divisible[n_, k_]}}], arg2,

opts:OptionsPattern[]] :=

GeneratingFunction[Piecewise[{{expr, Mod[n, k] == 0}}], n, x, opts];

(* "multisection formula" *)

GeneratingFunction[Piecewise[{{expr_, Mod[n_, k_] == j_}}], arg2,

opts:OptionsPattern[]] /; FreeQ[k, n] :=

With[{var = Unique[]},

k^(-1) * Sum[Exp[-var*j*2*Pi*I/k] * GeneratingFunction[

expr, n, Exp[var * 2 * Pi *I /k] * x, opts],

{var, 0, k-1}]

];

Protect[Evaluate[protected]];

End[]

EndPackage[]

SymbolicMethod.m

(* Mathematica Package *)

BeginPackage["Genfunlib`SymbolicMethod`"]

Spec::usage = "Spec[{lhs1 == rhs1, ...}, labeled] is a combinatorial " <>

"specification.";

SMPlus::usage = "SMPlus is the addition construction from the symbolic method.";

SMTimes::usage = "SMTimes is the multiplication construction from the " <>

"symbolic method.";

SMSeq::usage = "SMSeq is the sequence construction from the symbolic method.";

SMCyc::usage = "SMCyc is the cycle construction from the symbolic method.";

SMSet::usage = "SMSet is the set construction from the symbolic method.";

SMPointing::usage = "SMPointing is the pointing construction from the " <>

"symbolic method.";

SMSub::usage = "SMSub is the substitution construction from the symbolic " <>

"method.";

SMMultiset::usage = "SMMultiset is the multiset construction from the " <>

"symbolic method.";

Cardinality::usage = "Cardinality is an option for symbolic method " <>

"constructions which takes a Function on the integers with which to " <>

"restrict the construction.";

Restricted::usage = "Restricted[class, pred] is the class of objects from " <>

"class with sizes in pred^(-1)(True).";

ZClass::usage = "ZClass[n] is the nth atomic class.";

EClass::usage = "EClass is the epsilon class.";

ToGenfunlibSpec::usage = "ToGenfunlibSpec[mapleString, labeled] converts " <>

"from Combstruct to Genfunlib combinatorial specification syntax.";

ToGFEqns::usage = "ToGFEqns[spec, indeterminate] gives the system of power " <>

"series equations corresponding to spec, using the given indeterminate."

Begin["`Private`"] (* Begin Private Context *)

(* ::Section:: *)

(* Syntactic input validation *)

Spec::invalid = "Invalid specification.";

validatePred[_Function] := True;

validatePred[___] := False;

validateRHS[sym_, nonTerms_] /; MemberQ[nonTerms, sym] := True;

validateRHS[EClass, _] := True;

validateRHS[ZClass[n_Integer?Positive], _] := True;

validateRHS[SMPlus[args__], nonTerms_] := And @@ (validateRHS[#, nonTerms]& /@

{args});

validateRHS[SMTimes[args__], nonTerms_] := And @@ (validateRHS[#, nonTerms]& /@

{args});

validateRHS[SMSeq[arg_], nonTerms_] := validateRHS[arg, nonTerms];

validateRHS[SMSeq[arg_, Cardinality -> pred_], nonTerms_] := validateRHS[arg,

nonTerms] && validatePred[pred];

validateRHS[SMCyc[arg_], nonTerms_] := validateRHS[arg, nonTerms];

validateRHS[SMCyc[arg_, Cardinality -> pred_], nonTerms_] := validateRHS[arg,

nonTerms] && validatePred[pred];

validateRHS[SMSet[arg_], nonTerms_] := validateRHS[arg, nonTerms];

validateRHS[SMSet[arg_, Cardinality -> pred_], nonTerms_] := validateRHS[arg,

nonTerms] && validatePred[pred];

validateRHS[SMMultiset[arg_], nonTerms_] := validateRHS[arg, nonTerms];

validateRHS[SMMultiset[arg_, Cardinality -> pred_], nonTerms_] := validateRHS[

arg, nonTerms] && validatePred[pred];

validateRHS[SMSub[first_, second_], nonTerms_] := validateRHS[first,

nonTerms] && validateRHS[second, nonTerms];

validateRHS[SMPointing[arg_], nonTerms_] := validateRHS[arg, nonTerms];

validateRHS[Restricted[arg_, pred_], nonTerms_] := validatePred[pred] &&

validateRHS[arg, nonTerms];

validateRHS[___] := False;

validateSpecSyntax[spec:Spec[list:{HoldPattern[_ == _]..},

labeled:True|False]] := Module[

{

ok = True,

lhss = list[[All, 1]],

rhss = list[[All, 2]]

},

ok = ok && MatchQ[lhss, {(_Symbol|(_Symbol)[_Integer])...}];

ok = ok && Length@lhss == Length@Union@lhss;

If[labeled && MemberQ[list, SMMultiset, {0, Infinity}, Heads -> True],

ok = False];

ok = ok && And @@ (validateRHS[#, lhss]& /@ rhss);

If[ !ok, Message[Spec::invalid]];

ok

];

validateSpecSyntax[___] := False;

(* ::Section:: *)

(* To GF eqns *)

(* the function restrictedSum returns a simplified form of *)

(* Sum[generalTerm*Boole[pred[n]], {n, slb, sub}] *)

restrictedSum[generalTerm_, Function[GreaterEqual[ub_: Infinity, Slot[1],

lb_: 0]], {n_, slb_, sub_}] :=

Sum[generalTerm, {n, Max[slb, lb], Min[sub, ub]}];

restrictedSum[generalTerm_, Function[LessEqual[lb_: 0, Slot[1], ub_: Infinity]],

{n_, slb_, sub_}] :=

Sum[generalTerm, {n, Max[slb, lb], Min[sub, ub]}];

restrictedSum[generalTerm_, pred_, {n_, slb_, sub_}] :=

Sum[Boole[pred[n]] * generalTerm, {n, slb, sub}];

(* labeled *)

ToGFEqns[ spec:Spec[list_List, True], indet_Symbol ] := Module[

{

numAtomicClasses = Max @@ First /@ Cases[list, ZClass[n_], Infinity] //

Max[0,#]&,

nonTerms = list[[All, 1]],

indets,

ret = Hold[list]

},

(

indets = Sequence @@ indet /@ Range[numAtomicClasses];

ret = Replace[ret,

{

ZClass[n_] :> indet[n],

EClass :> 1,

sym_Symbol?(MemberQ[nonTerms, #1] &) :> sym[indets]

},

{1, Infinity}, Heads -> True

];

ret = FixedPoint[Function[iter, Replace[iter,

{

SMPlus[args__] :> Plus[args],

SMTimes[args__] :> Times[args],

SMSeq[arg_] :> 1 / (1 - arg),

SMSeq[arg_, Cardinality -> pred_] :> With[{unique = Unique[]},

restrictedSum[arg^unique, pred, {unique, 0, Infinity}]],

SMCyc[arg_] :> Log[ 1/(1 - arg) ],

SMCyc[arg_, Cardinality -> pred_] :> With[{unique = Unique[]},

Boole[pred[0]] + restrictedSum[arg^unique / unique, pred, {

unique, 1, Infinity}]],

SMSet[arg_] :> Exp[arg],

SMSet[arg_, Cardinality -> pred_] :> With[{unique = Unique[]},

restrictedSum[arg^unique / (unique!), pred, {unique, 0,

Infinity}]],

Restricted[expr_, {}] :> expr,

Restricted[expr_, pred_] :> With[{unique = Unique[]},

restrictedSum[

SeriesCoefficient[expr, {indet[1], 0, unique}] * indet[1]^

unique,

pred, {unique, 0, Infinity}

]],

SMPointing[expr_] :> indet[1] * D[expr, indet[1]],

SMSub[func_, arg_] :> (func /. indet[1] :> arg)

}, {1, Infinity}, Heads -> True

]], ret];

ReleaseHold[ret]

) /; validateSpecSyntax[spec]

];

(* unlabeled *)

ToGFEqns[spec:Spec[list_List, False], indet_Symbol] := Module[

{

numAtomicClasses = Max @@ First /@ Cases[list, ZClass[n_], Infinity] //

Max[0,#]&,

nonTerms = list[[All, 1]],

indets,

unique, uniqueAux,

ret = Hold[list]

},

(

indets = Sequence @@ indet /@ Range[numAtomicClasses];

ret = Replace[ret,

{

ZClass[n_] :> indet[n],

EClass :> 1,

sym_Symbol?(MemberQ[nonTerms, #1] &) :> sym[indets]

},

{1, Infinity}, Heads -> True

];

ret = FixedPoint[Function[iter, Replace[iter,

{

SMPlus[args__] :> Plus[args],

SMTimes[args__] :> Times[args],

SMSeq[arg_] :> 1 / (1 - arg),

SMSeq[arg_, Cardinality -> pred_] :> With[{unique = Unique[]},

restrictedSum[arg^unique, pred, {unique, 0, Infinity}]],

157



SMCyc[arg_] :> With[{unique = Unique[]},

Sum[EulerPhi[unique]/unique * Log[1/(1 - (arg /. indet[n_] :>

indet[n]^unique))], {unique, 1, Infinity}]],

(* p. 730 of Flajolet and Sedgewick *)

SMCyc[arg_, Cardinality -> pred_] :> With[{unique = Unique[],

uniqueAux = Unique[]},

restrictedSum[

SeriesCoefficient[

Sum[EulerPhi[unique]/unique * Log[1/(1 - uniqueAux^

unique * (arg /. indet[n_] :> indet[n]^unique))],

{unique, 1, Infinity}

],

{uniqueAux, 0, unique}

],

pred,

{unique, 0, Infinity}

]

],

SMSet[arg_] :> With[{unique = Unique[]},

Exp@Sum[(-1)^(unique - 1)/unique * (arg /. indet[n_] :>

indet[n]^unique), {unique, 1, Infinity}]],

SMSet[arg_, Cardinality -> pred_] :> With[{unique = Unique[],

uniqueAux = Unique[]},

restrictedSum[

SeriesCoefficient[

Exp@Sum[(-1)^(unique - 1)/unique * uniqueAux^unique *

(arg /. indet[n_] :> indet[n]^unique),

{unique, 1, Infinity}

],

{uniqueAux, 0, unique}

],

pred,

{unique, 0, Infinity}

]

],

SMMultiset[arg_] :> With[{unique = Unique[]},

Exp@Sum[1/unique * (arg /. indet[n_] :> indet[n]^unique),

{unique, 1, Infinity}]],

SMMultiset[arg_, Cardinality -> pred_] :> With[{unique = Unique[],

uniqueAux = Unique[]},

restrictedSum[

SeriesCoefficient[

Exp@Sum[1/unique * uniqueAux^unique * (arg /.

indet[n_] :> indet[n]^unique),

{unique, 1, Infinity}

],

{uniqueAux, 0, unique}

],

pred,

{unique, 0, Infinity}

]

],

Restricted[expr_, {}] :> expr,

Restricted[expr_, param_] :> With[{unique = Unique[]},

restrictedSum[

SeriesCoefficient[expr, {indet[1], 0, unique}] *

indet[1]^unique,

pred, {unique, 0, Infinity}

]],

SMPointing[expr_] :> indet[1] * D[expr, indet[1]],

SMSub[func_, arg_] :> (func /. indet[1] :> arg)

}, {1, Infinity}, Heads -> True

]], ret];

ReleaseHold[ret]

) /; validateSpecSyntax[spec]

];

(* ::Section:: *)

(* Combstruct grammar to Genfunlib spec *)

(* this function assumes there are no parentheses except to group function *)

(* arguments, assumes nonterminals are only one character, and *)

(* assumes the grammar is explicit *)

ToGenfunlibSpec[str_String, labeled:(True|False)] := Module[

{

ret = str

},

ret = StringReplace[ret, c_?UpperCaseQ :> ToLowerCase[c] <> ToLowerCase[c]];

ret = StringReplace[ret, {

" = " -> " == ",

"(" -> "[", ")" -> "]",

"pprod" -> "SMTimes",

"uunion" -> "SMPlus",

"sset" -> "SMSet",

"ppowersset" -> "SMSet",

"ccycle" -> "SMCyc",

"ssubst" -> "SMSub",

"ssequence" -> "SMSeq",

"zz" -> "ZClass[1]",

"aatom" -> "ZClass[1]",

"eepsilon" -> "EClass"

}];

ret = ToExpression[ret, InputForm, Hold];

ret = Replace[ret,

SMSub[first_, second_] :> SMSub[second, first], {0, Infinity}];

ret = Replace[ret, {

(head : (SMSet | SMSeq | SMCyc))[arg_, rel_[Global`card, k_]] :>

head[arg, Cardinality -> (rel[#, k] &)],

(head : (SMSet | SMSeq | SMCyc))[arg_, rel_[k_, Global`card]] :>

head[arg, Cardinality -> (rel[k, #] &)]

},

{0, Infinity}];

Spec[ret//ReleaseHold, labeled]

];

ToGFEqns::invalidArgumentSyntax = "Invalid argument syntax.";

ToGenfunlibSpec::invalidArgumentSyntax = "Invalid argument syntax.";

ToGFEqns[___] /; (Message[ToGFEqns::invalidArgumentSyntax]; False) := Null;

ToGenfunlibSpec[___] /; (Message[ToGenfunlibSpec::invalidArgumentSyntax];

False) := Null;

End[] (* End Private Context *)

EndPackage[]

